
 LINQدورة 
 

 بسم الله الرحمن الرحيم
 

 يقذية
المختلفة  البٌانات أنواع من العدٌد مع تتعامل عندما باٌزٌك بإمكانٌات بسٌطة وقوٌة لفٌجول استعلامٌة إمكانٌات Linq تضٌؾ 

 تقوم التً البٌانات أنواع من نوع لكل للاستعلام مختلفة صٌؽة مع العمل أو معالجته تتم كً بٌانات قاعدة إلى الاستعلام إرسال إلى فبالإضافة

. تستخدمه الذي البٌانات نوع عن النظر بؽض موحدة صٌؽة مستخدمة باٌزٌك فٌجول لؽة من كجزء الاستعلامات Linq تقدم. عنها بالبحث

 ADO .net أو الذاكرة فً والمصفوفات المجموعات أو Xml أو SQL Server بٌانات قاعدة من البٌانات عن الاستعلام من تمكنك وهً

Datasets  الأمر الذي ٌجعلها قادرة على الاستعلام من أي قاعدة بٌانات ٌمكن ربطها معDataSetٌدعم بعٌد أو محلً بٌانات مصدر أي  أو 

Linq ونتائج باٌزٌك فٌجول بلؽة مكتوبة أصبحت استعلاماتك لأن الشائعة باٌزٌك فٌجول لؽة عناصر باستخدام كله ذلك عمل ٌمكنك حٌث 

 عند الاستعلامات فً للأخطاء واكتشافك أسرع للكود كتابتك ٌجعل مما IntelliSense داعمة قوٌة بٌانات أنواع كؤؼراض تعاد الاستعلام

 كما. البحث فً الدقة من لمزٌد إضافٌة لاستعلامات مصدرا تكون أن ٌمكن Linq استعلامات أن كما التنفٌذ وقت من بدلا المشروع ترجمة

 .بسهولة استعلامك نتائج وتعدٌل استعراض من المستخدم ممكنا التحكمات مع ربطها ٌمكن

 وهذا مثال عن استعلام بسٌط ٌعٌد قائمة شركات الزبائن الموجودٌن فً إٌطالٌا

Dim itaCus = From cus In NwDs.Customers _ 

             Where cus.Country = "Italy" _ 

             Select cus.ContactName, cus.CompanyName 

 

دعنا لا نقلق الآن بخصوص صٌؽة الاستعلام على كل حال إن كنت متمكنا من كتابة استعلامات سٌكول سٌرفر لن تجد صعوبة فً فهم 

 صٌؽتها بما أن الصٌؽة مشابهة مع بعض الاختلافات طبعا والاستعلام السابق ٌماثل استعلام سٌكول سٌرفر التالً

SELECT COMPANYNAME FROM CUSTOMERS WHERE COUNTRY='Italy' 

  أكثر تعقٌد فالكود التالً ٌعٌد قائمة بالزبائن وٌعٌد تجمٌعهم حسب الموقعLinqكما ٌمكن أن ٌكون استعلام 

Dim customers As List(Of Customer) = GetCustomerList() 

Dim customersByCountry = From cust In customers _ 

                         Order By cust.Country, cust.City _ 

                         Group By CountryName = cust.Country _ 

                         Into RegionalCustomers = Group, Count() _ 

                         Order By CountryName 

For Each country In customersByCountry 

     Console.WriteLine(country.CountryName & _ 

           " (" & country.Count & ")" & vbCrLf) 

    For Each customer In country.RegionalCustomers 

         Console.WriteLine(vbTab & customer.CompanyName & _ 

               " (" & customer.City & ")") 

    Next 

Next 

  مباشرةDataGrid فً برنامجنا بعدة أشكال فالمثال الأول مثلا ٌمكننا عرض نتٌجته فً Linqحٌث ٌمكننا استخدام استعلام 

Me.DataGridView1.DataSource = itaCus.ToList 

 مثلا واستخدام النتائج ففً المثال التالً نستخدم الاستعلام الوارد بالمثال الأول لإظهار قائمة For…Eachأو ٌمكننا إدخاله ضمن حلقة 

 ListBoxالشركات فً 

Me.ListBox1.Items.Clear() 

For Each cust In itaCus 

       Me.ListBox1.Items.Add(cust.CompanyName) 

Next 

 

 

 Linq -  LINQ Providersيزودات 

 

 ٌؤخذ المزود Linq فً فٌجول باٌزٌك بحسب مصدر البٌانات الذي تستخدمه فعندما تكتب استعلام Linq بتنظٌم استعلامات Linqٌقوم مزود 

ذلك الاستعلام وٌترجمه إلى أوامر ٌستطٌع مصدر البٌانات تنفٌذها وٌقوم أٌضا بتحوٌل البٌانات من الأؼراض المصدرٌة لٌشكل نتائج 
 : التالٌةLinqوٌضم فٌجول باٌزٌك مزودات . الاستعلام وأخٌرا ٌقوم بتحوٌل الأؼراض إلى بٌانات عندما ترسل التحدٌثات للمصدر

 Linq to Objects ٌمكنك هذا المزود من الاستعلام فً المجموعات والمصفوفات فً الذاكرة إذا كانت ؼرضك ٌدعم الواجهة 

IEnumerable أو الواجهة IEnumerable(T) بحٌث ٌمكنك المزود من الاستعلام عنها وٌمكنك تمكٌن هذا المزود باستٌراد 

  والذي ٌكون مستوردا بشكل افتراضً فً مشارٌع فٌجول باٌزٌكSystem.Linqالمجال 

 Linq to SQL ٌمكنك هذا المزود من الاستعلام من قواعد بٌانات SQL Server والتحدٌث إلٌها وٌجعل من السهل ربط أؼراض 

 Object Relational بتقدٌم Linq to SQLوٌسهل فٌجول باٌزٌك العمل مع . التطبٌق مع الجداول والأؼراض فً قواعد البٌانات
Designer والذي ٌمكنك من إنشاء Object Model فً التطبٌق ٌرتبط مع الأؼراض فً قاعدة البٌانات وٌقدم الـ O/R 
Designer إمكانٌة التعامل مع الإجراءات والوظائؾ المخزنة وإجراءات الؽرض DataContext الذي ٌدٌر الاتصال مع قاعدة 

 البٌانات وٌخزن الحالة من أجل تصادم البٌانات التفاإلً
 Linq to Xml ٌمكنك من الاستعلام من Xml والتعدٌل علٌها بحٌث ٌمكنك تعدٌل محتوٌات Xml الموجودة فً الذاكرة أو ٌمكنك 

  أو حفظه Xmlتحمٌل ملؾ 

 Linq to Dataset ٌمكنك من الاستعلام من ADO .net Datasets والتعدٌل علٌها وإضافة قوة Linq للتطبٌقات التً تستخدم 

Datasets تسهل وتوسع إمكانٌات الاستعلام والتجمٌع والتحدٌث فً الـ Datasetفً تطبٌقك  

  Linqبنية اسحعلايبت 

 
 بتعبٌر الاستعلام وهو ٌتؤلؾ من تولٌفة من تراكٌب الاستعلام التً تحدد مصدر البٌانات ومتؽٌرات Linq ٌشار عادة إلى استعلام 

التكرار الخاصة بالاستعلام كما ٌمكنه أن ٌتضمن تعلٌمات من أجل الفرز أو التصفٌة أو التجمٌع أو الضم أو الحساب لٌتم تطبٌقها على 
 . ولهذا ستجد أن معظم الصٌؽة مؤلوفةSQLالبٌانات المصدرٌة وصٌؽتها تكون مشابهة لصٌؽة الـ 

 الذي ٌحدد مصدر البٌانات والمتؽٌرات التً تشٌر إلى كل عنصر من البٌانات المصدرٌة بشكل مستقل وهً تدعى Fromٌبدأ الاستعلام بقسم 

 حٌث ٌكون قسم Aggregate مطلوب من أجل الاستعلام إلا فً استعلامات التجمٌع Fromبمتؽٌرات المجال أو متؽٌرات التكرار وقسم 

From فٌها اختٌاري وبعد تعرٌؾ مجال ومصدر الاستعلام فً قسم From أو فً قسم Aggregate ٌمكنك تضمٌن أي تركٌب من أقسام 

 cust ومتؽٌر التكرار Customersفمثلا الاستعلام التالً ٌحدد مصدر مجموعة من بٌانات الزبائن بالمتؽٌر . الاستعلام

Dim queryResults = From cust In customers _ 

                   Select cust.CompanyName 

 

وهذا المثال ٌشكل استعلام مقبول بذات نفسه ومع ذلك ٌصبح الاستعلام أقوى عندما تضٌؾ أقسام استعلام أخرى لتحدٌد النتائج فمثلا ٌمكنك 
 لتصفٌة النتائج إلى قٌمة أو أكثر وتكون تعابٌر الاستعلام عبارة عن سطر واحد من الكود بحٌث ٌمكنك إضافة أقسام Whereإضافة قسم 

وٌمثل الكود التالً _ استعلام جدٌدة لنهاٌة الاستعلام كما ٌمكنك فصل الاستعلام إلى عدة أسطر لتحسٌن قراءة كودك باستخدام المحرؾ 

 Whereاستعلاما ٌستخدم قسم 

Dim queryResults = From cust In customers _ 

                   Where cust.Country = "USA" 

 

 Linq قسم قوي آخر فً الاستعلام حٌث ٌمكنك من إعادة الحقول المختارة فقط من مصدر البٌانات وتعٌد استعلامات selectوٌمثل قسم 

 للعودة بحقل selectوٌمكن استخدام قسم . مجموعة تعدادٌة من الأؼراض القوٌة النوع كما ٌمكنها إعادة أنواع مجهولة أو أنواع معروفة

 مجموعة Selectوعندما ٌعٌد قسم . واحد فقط من مصدر البٌانات وعندما تفعل هذا ٌكون نوع المجموعة المعادة هو نوع بٌانات ذلك الحقل

من الحقول من مصدر البٌانات تكون المجموعة المعادة من النوع المجهول وٌمكنك مطابقة الحقول المعادة من الاستعلام مع حقول من نوع 
 معروؾ محدد وٌظهر الكود التالً تعبٌر استعلام ٌعٌد مجموعة نوعها مجهول تضم أرقاما مع بٌانات من الحقل المحدد من مصدر البٌانات

Dim queryResults = From cust In customers _ 

                   Where cust.Country = "USA" _ 

                   Select cust.CompanyName, cust.Country 

 

 واحد أو أكثر أو From لدمج عدة مصادر من البٌانات فً نتٌجة واحدة حٌث ٌمكن عمل هذا باستخدام قسم Linqٌمكن استخدام استعلامات 

 وٌعٌد مجموعة من نوع Order و Customer وٌظهر الكود التالً تعبٌر استعلام ٌضم بٌانات Group Join أو Joinباستخدام أقسام 

 Order و Customerمجهول تحتوي بٌانات من 

Dim queryResults = From cust In customers, ord In orders _ 

                   Where cust.CustomerID = ord.CustomerID _ 

                   Select cust, ord 

 

 ٌمتلك Customer وكل ؼرض Customer لبناء استعلامات شجرٌة تحتوي مجموعة من أؼراض Group Joinٌمكنك استخدام قسم 

 و Customerوٌظهر المثال التالً تعبٌر استعلام ٌدمج بٌانات .  لذلك الزبونorderخاصٌة تحتوي مجموعة تتضمن جمٌع أؼراض 

Order كنتٌجة شجرٌة وٌعٌد مجموعة من نوع مجهول وٌعٌد الاستعلام نوعا ٌتضمن الخاصٌة CustomerOrders تحتوي على مجموعة 

 تحتوي على مجموع إجمالً الطلبات OrderTotal وتتضمن أٌضا الخاصٌة Customer وبٌانات Orderتحتوي على مجموعة من بٌانات 

 لذلك الزبون

Dim queryResults = From cust In customers _ 

                   Group Join ord In orders On _ 

                   cust.CustomerID Equals ord.CustomerID _ 

                   Into CustomerOrders = Group, _ 

                        OrderTotal = Sum(ord.Total) _ 

                   Select cust.CompanyName, cust.CustomerID, _ 

                          CustomerOrders, OrderTotal 

 Linq - Visual Basic LINQ Query Operatorsيعبيلات اسحعلاو 

 طرائق ٌمكنك استدعاإها لإنشاء الاستعلامات وتولٌها Linq والمجالات التً تدعم System.Linq تتضمن الفئات فً المجال 

 بحسب حاجة التطبٌق وٌتضمن فٌجول باٌزٌك كلمات مفتاحٌة لأقسام الاستعلام الشائعة

From Clause 

  المجموعة المصدر أو متؽٌر التكرار للاستعلامFrom وٌحدد قسم Aggregate أو Fromٌجب أن ٌبدأ الاستعلام بقسم 

' Returns the company name for all customers for whom 

' State is equal to "WA". 

Dim names = From cust In customers _ 

            Where cust.State = "WA" _ 

            Select cust.CompanyName 

 

Select Clause 

 اختٌاري ٌحدد مجموعة من متؽٌرات التكرار للاستعلام

' Returns the company name and ID value for each 

' customer as a collection of a new anonymous type. 

Dim customerList = From cust In customers _ 

                   Select cust.CompanyName, cust.ID 

 

 Aggregate أو From موجودا فً الاستعلام فتتؤلؾ متؽٌرات التكرار للاستعلام من تلك المحددة فً قسم Selectو إن لم ٌكن قسم 

Where Clause 

 اختٌاري وٌحدد شرط التصفٌة للاستعلام

' Returns all product names for which the Category of 

' the product is "Beverages". 

Dim names = From product In products _ 

            Where product.Category = "Beverages" _ 

            Select product.Name 

 

Order By Clause 

 اختٌاري وٌحدد ترتٌب الفرز للأعمدة فً الاستعلام

' Returns a list of books sorted by price in  

' ascending order. 

Dim titlesAscendingPrice = From b In Books _ 

                           Order By b.Price 

Join Clause 

 اختٌاري وٌدمج مجموعتٌن ضمن مجموعة واحدة

Dim Intrst = From i In DsDesposits.InterestRates_  

             Join d In DsDesposits.Deposits_  

             On i.InterestID Equals d.InterestID_  

             Select i.InterestID, i.InterestRate, i.DepositPreiod 

Group by Clause 

 اختٌاري وٌقوم بتجمٌع عناصر نتٌجة الاستعلام وٌمكن استعماله لتطبٌق إجراءات تجمٌع لكل مجموعة

' Returns a list of orders grouped by the order date 

' and sorted in ascending order by the order date. 

Dim orders = From order In orderList _ 

             Order By order.OrderDate _ 

             Group By OrderDate = order.OrderDate _ 

             Into OrdersByDate = Group 

Group Join Clause 

 اختٌاري وٌجمع مجموعتٌن ضمن مجموعة شجرٌة واحدة

' Returns a combined collection of customers and 

' customer orders. 

Dim customerList = From cust In customers _ 

                   Group Join ord In orders On _ 

                   cust.CustomerID Equals ord.CustomerID _ 

                   Into CustomerOrders = Group, _ 

                        OrderTotal = Sum(ord.Total) _ 

                   Select cust.CompanyName, cust.CustomerID, _ 

                          CustomerOrders, OrderTotal 

 

Aggregate Clause 

 ٌطبق واحدة أو أكثر من وظائؾ التجمٌع على المجموعة Aggregate وقسم Aggregate أو قسم Fromٌجب بدء الاستعلام دوما إما بقسم 

  لحساب مجموع جمٌع العناصر المعادة بالاستعلامAggregateفمثلا ٌمكن استخدام قسم 

' Returns the sum of all order totals. 

Dim orderTotal = Aggregate order In Orders _  

                 Into Sum(order.Total) 

 

  لإجراء عملٌة حسابٌة على مجموعة استعلامAggregate لتعدٌل الاستعلام فمثلا ٌمكن استخدام قسم Aggregateكما ٌمكنك استخدام قسم 

' Returns the customer company name and largest  

' order total for each customer. 

Dim customerMax = From cust In customers _ 

                  Aggregate order In cust.Orders _ 

                  Into MaxOrder = Max(order.Total) _ 

                  Select cust.CompanyName, MaxOrder 

 

Let Clause 

 اختٌاري وٌقوم بحساب قٌمة وٌضعها فً متؽٌر جدٌد

' Returns a list of products with a calculation of 

' a ten percent discount. 

Dim discountedProducts = From prod In products _ 

                         Let Discount = prod.UnitPrice * 0.1 _ 

                         Where Discount >= 50 _ 

                         Select prod.ProductName, prod.UnitPrice, Discount 

Distinct Clause 

 اختٌاري وهو ٌضبط القٌم المعادة من الاستعلام بحٌث لا ٌجلب قٌما مكررة

' Returns a list of cities with no duplicate entries. 

Dim cities = From item In Customers _ 

             Select customer.City_ 

             Distinct 

 

Skip Clause 

 اختٌاري ٌتجاوز عددا معٌنا من العناصر فً المجموعة وٌعٌد الباقً

' Returns a list of customers. The first 10 customers 

' are ignored and the remaining customers are 

' returned. 

Dim customerList = From cust In customers _ 

                   Skip 10 

 



9 

Skip While Clause 

  ثم ٌعٌد باقً العناصرTrueاختٌاري ٌتجاوز عناصر المجموعة طالما قٌمة الشرط 

' Returns a list of customers. The query ignores all 

' customers until the first customer for whom 

' IsSubscriber returns false. That customer and all 

' remaining customers are returned. 

Dim customerList = From cust In customers _ 

                   Skip While IsSubscriber(cust) 

 

Take Clause 

 اختٌاري وٌعٌد عددا من العناصر المتجاورة فً بداٌة المجموعة

' Returns the first 10 customers. 

Dim customerList = From cust In customers _ 

                   Take 10 

 

Take While Clause 

  وٌتجاهل بقٌة العناصرTrueاختٌاري ٌقوم بتضمٌن عناصر المجموعة طالما قٌمة الشرط 

' Returns a list of customers. The query returns 

' customers until the first customer for whom  

' HasOrders returns false. That customer and all  

' remaining customers are ignored. 

Dim customersWithOrders = From cust In customers _ 

                          Order By cust.Orders.Count Descending _ 

                          Take While HasOrders(cust) 

 

 حٌث ٌمكنك Linq باستدعاء عناصر المجموعات والأنواع المستعلم عنها التً ٌوفرها Linqكما ٌمكنك استخدام خصائص إضافٌة لاستعلام 

 لدمج ناتج Unionاستخدام هذه الإمكانٌات الإضافٌة باستدعاء معامل استعلام على نتٌجة الاستعلام فمثلا الكود التالً ٌستخدم الطرٌقة 

  لإعادة ناتج الاستعلام كقائمةToList(TSource)استعلامٌن فً نتٌجة استعلام واحدة وٌستخدم الطرٌقة 

Public Function GetAllCustomers() As List(Of Customer) 

    Dim customers1 = From cust In domesticCustomers 

    Dim customers2 = From cust In internationalCustomers 

 

    Dim customerList = customers1.Union(customers2) 

    Return customerList.ToList() 

End Function 

  لابذ ين الإطلاع عهيهبLinqيىاضيع يحعهقة بحقنية 

 

 Local Type Inferenceالاسحذلال انًحهي عهى اننىع 

 

 

 لتحدٌد نوع المتؽٌرات المحلٌة التً تم Type Inference الاستدلال على النوع 2008 ٌستخدم المترجم فً فٌجول باٌزٌك 

 فً تعبٌر التصرٌح حٌث ٌستدل المترجم على نوع المتؽٌر من نوع التعبٌر الذي ٌضبط قٌمة ذلك Asالتصرٌح عنها بدون استخدام فقرة 

 المتؽٌر مما ٌوفر إمكانٌة تعرٌؾ المؽٌرات بدون تحدٌد نوعها كما فً المثال التالً

Public Sub inferenceExample() 

 

    ' Using explicit typing . 

    Dim num1 As Integer = 3 

 

    ' Using local type inference. 

    Dim num2 = 3 

 

End Sub 

 

 فً المثال السابق حقلا فً فئة بدلا عن num2 فإن كان Class Fieldsولا ٌمكن استخدام الاستدلال على النوع عند تعرٌؾ الحقول فً الفئة 

 إن كان Object على أنه ؼرض num2 وسوؾ ٌصنؾ Option Strict Onكونه متؽٌرا محلٌا فسوؾ ٌولد التصرٌح خطؤ إذا كان 

Option Strict Off وبشكل مشابه فنوع المتؽٌرات الساكنة Static Variables لا ٌمكن الاستدلال علٌها إن كان Option Strict On وإن 

 فً المثال السابق أن ٌكون من num2 فإن لم تكن ترٌد من المتؽٌر Object فنوع المتؽٌر الساكن سٌكون ؼرض Option Strict Offكان 

  فٌمكنك تحدٌد نوعا آخر عند التصرٌح عنهIntegerالنوع 

Dim num3 As Object = 3 or Dim num4 As Double = 3 
 

 الذي سٌكون نوعه معروفا فقط فً زمن Late Bindingوالكود الذي ٌستخدم استدلال النوع ٌشابه الكود الذي ٌعتمد على الربط المتؤخر 

ومعرفة النوع بشكل مبكر ٌمكن المترجم من تحدٌد المشاكل قبل التنفٌذ وحجز الذاكرة بدقة وإجراء عملٌات التحسٌن الأخرى . التشؽٌل

 والمساعدة حول أعضاء ذلك الؽرض بالإضافة إلى تفضٌله لاعتبارات IntelliSenseبالإضافة إلى تمكٌن بٌئة التطوٌر من تزوٌد المبرمج بـ 

 والوصول إلى الأعضاء Objectخاصة بالأداء بسبب أن جمٌع البٌانات التً تخزن باستخدام الربط المتؤخر ٌجب تؽلٌفها وكؤنها من النوع 

 .فً زمن التشؽٌل سٌكون أبطؤ

 فً تعبٌر التصرٌح وضبط قٌمة لذلك المتؽٌر فٌستخدم Asٌحدث الاستدلال على النوع عندما ٌتم التصرٌح عن المتؽٌر بدون استخدام فقرة 

 Stringالمترجم نوع تلك القٌمة كنوع للمتؽٌر فمثلا سطور الكود التالٌة تعرؾ متؽٌرا من النوع 

' Using explicit typing. 

Dim name1 As String = "Springfield" 

 

' Using local type inference. 

Dim name2 = "Springfield" 

 

 Integerوٌستعرض الكود التالً طرٌقتان متكافئتان لإنشاء مصفوفة من النوع 

' Using explicit typing. 

Dim someNumbers1() As Integer = New Integer() {4, 18, 11, 9, 8, 0, 5} 

 

' Using local type inference. 

Dim someNumbers2 = New Integer() {4, 18, 11, 9, 8, 0, 5} 

 بؤنه من numكما ٌمكنك استخدام الاستدلال على النوع لتحدٌد نوع متؽٌر التحكم لحلقة تكرارٌة ففً الكود التالً سٌتعرؾ المترجم على 

 Integer عبارة عن مصفوفة someNumbers2 لأن Integerالنوع 

Dim total = 0 

For Each number In someNumbers2 

    total += number 

Next 

 

  أٌضا لتحدٌد نوع اسم المصدر كما هو واضح فً المثال التالUsingًوٌستخدم الاستدلال على النوع فً العبارة 

Using proc = New System.Diagnostics.Process 

    ' Insert code to work with the resource. 

End Using 

 

 Lists عبارة عن pList2 و pList1وٌستدل على نوع المتؽٌر من القٌمة المعادة من الإجراء أٌضا كما هو ظاهر فً الكود التالً حٌث ٌكون 

of Processes 

' Using explicit typing. 

Dim pList1() As Process = Process.GetProcesses() 

 

' Using local type inference. 

Dim pList2 = Process.GetProcesses() 

 

 ٌمكنك من تحدٌد إذا كان الاستدلال المحلً على النوع مسموحا أم لا فً ملؾ Option Infer خٌارا جدٌدا هو 2008وقد قدم فٌجول باٌزٌك 

 فلتمكٌن أو تعطٌل خٌار الاستدلال على النوع اكتب التعبٌر المناسب من السطرٌن التالٌٌن فً بداٌة الملؾ. معٌن

Option Infer On 

Option Infer Off 

 

 من أجل المشارٌع التً Option Infer On فً الكود فالمترجم سٌستخدم الخٌار الافتراضً Option Inferوإن لم تقم بتحدٌد قٌمة للخٌار 

وإن تضاربت .  من أجل المشارٌع التً تمت ترقٌتها من إصدارات سابقةOption Infer Off والخٌار Visual Basic 2008تم إنشاإها فً 

 فً الملؾ مع القٌمة المضبوطة فً خٌارات بٌئة التطوٌر أو فً سطر الأوامر فسوؾ ٌتم استخدام القٌمة Option Inferقٌمة الخٌار 

 .الموجودة فً الملؾ

 أو Class Fields ولا ٌمكن استخدامها فً تعرٌؾ حقول الفئة Non-Staticوٌستخدم الاستدلال على النوع فقط فً المتؽٌرات الؽٌر ساكنة 

  Functions أو الإجراءات Propertiesالخصائص 

 Anonymous Types انًجهىنة الأنىاع

 كتابة بدون Objects الأؼراض إنشاء من تمكنك والتً anonymous types المجهولة الأنواع 2008 ستودٌو فٌجول  ٌقدم

 للاستخدام قابلا اسما للفئة ٌكون ولن أجلك من الفئة المترجم ٌولد ذلك عن وعوضا البٌانات نوع أجل من Class definition فئة تعرٌؾ

 لم البٌانات نوع أن وبما Object الؽرض تعرٌؾ عند تحددها التً الخصائص وتمتلك Object من مباشرة موروثة الفئات هذه تكون حٌث

 النوع من كمتؽٌر product المتؽٌر عن التالً المثال ٌصرح حٌث. anonymous type مجهول نوع أنه على إلٌه الإشارة ٌتم تحدٌده ٌتم

anonymous type الخاصٌتٌن ممتلكا Name  و Price 

' Variable product is an instance of a simple anonymous type. 
Dim product = New With {Key .Name = "paperclips", .Price = 1.29} 

 النتٌجة نوع تحدٌد لا ٌمكنك أنه وبما الاستعلام بواسطة المحددة البٌانات أعمدة لدمج المجهولة الأنواع التالً الاستعلام تعبٌر ٌستخدم حٌث
 من عدد ٌختار استعلام كتابة من المجهولة الأنواع فتمكنك معٌن استعلام ٌختارها أن ٌمكن التً بالأعمدة التنبإ إمكانٌة عدم بسبب مقدما

 ٌكون التالً المثال وفً. المعٌن الترتٌب بذلك المحددة الخصائص لتلك المماثل البٌانات نوع بإنشاء المترجم فٌقوم نرٌده ترتٌب بؤي الأعمدة
Products أؼراض من قائمة عن عبارة Product المتؽٌر ٌحمل بحٌث عدٌدة خصائص ٌمتلك منها وكل namePriceQuery تعرٌؾ 

 Price و Name الخاصٌتٌن تمتلك التً المجهولة الأنواع من مجموعة تنفٌذه عند ٌعٌد الذي الاستعلام

Dim namePriceQuery = From prod In products_  
                     Select prod.Name, prod.Price 

 خاصٌتٌن تمتلك التً المجهولة الأنواع من مجموعة تنفٌذه عند ٌعٌد الذي الاستعلام تعرٌؾ ٌحمل nameQuantityQuery  والمتؽٌر

Name و OnHand 

Dim nameQuantityQuery = From prod In products_  
                        Select prod.Name, prod.OnHand 

 

 Declaring an Anonymous Type مجهول نوع تعريف

 الإعلان عند فقط هذه الخصائص تحدٌد ٌمكنك بحٌث النوع ذلك خصائص لتحدٌد بناء قائمة ٌستخدم مجهول نوع من متؽٌر  تعرٌؾ
 Product1 ٌكون التالً المثال ففً المجهولة الأنواع فً والأحداث الطرائق مثل الفئات عناصر بقٌة استخدام ولا ٌمكن المجهول النوع عن

 Price و Name خاصٌتٌن ٌمتلك مجهول نوع من

 'Variable product1 is an instance of a simple anonymous type. 
Dim product1 = New With {.Name = "paperclips", .Price = 1.29} 

- 'or- 

 'product2 is an instance of an anonymous type with key properties. 
Dim product2 = New With {Key .Name = "paperclips", Key .Price = 1.29} 

 أم متساوٌٌن هما هل مجهولٌن نوعٌن لمقارنة استخدامها بإمكانك أصبح key properties  مفتاحٌة كخصائص الخصائص بتحدٌد قمت فإن

 نوع عن التصرٌح ٌماثل مجهول نوع عن التصرٌح أن ملاحظة مع. فقط للقراءة فهً تؽٌٌرها لا ٌمكن المفتاحٌة الخصائص فقٌم ذلك ومع لا
 object initializer الؽرض بانً باستخدام مسمى

 'Variable product3 is an instance of a class named Product. 
Dim product3 = New Product With {.Name = "paperclips", .Price = 1.29} 

 

 Key Properties انًفحبحية انخصبئص

 :أمور بعدة العادٌة عن المفتاحٌة الخصائص  تختلؾ

 مجهولٌن نوعٌن بٌن المساواة لمقارنة فقط المفتاحٌة الخصائص تستخدم 

 فقط للقراءة دائما فهً المفتاحٌة الخصائص قٌم تؽٌٌر لا ٌمكن 

 الـ ضمن تضمٌنها ٌتم المفتاحٌة الخصائص فقط Hash Code المجهولة الأنواع أجل من المترجم ٌولده الذي 

 Equality انًسبواة

 من كمتؽٌرٌن متؽٌرٌن بمعاملة المعالج وٌقوم المجهول النوع لنفس متؽٌرات تكون عندما متساوٌة المجهولة الأنواع متؽٌرات  تكون
 التالٌة الشروط فٌهما توفرت إذا النوع نفس

 المجمع نفس فً عنهما التصرٌح تم 

 الأحرؾ لحالة حساسة ؼٌر الأسماء مقارنة وتكون الترتٌب بنفس عنها التصرٌح وتم والنوع الاسم نفس خصائصهما تمتلك 

 أساسٌة كخصائص محددة فٌها الخصائص نفس 

 الأقل على واحدة أساسٌة خاصٌة نوع كل ٌمتلك 
 فقط لنفسه مساوٌا ٌكون مفتاحٌة خاصٌة أي الذي لا ٌمتلك مجهول نوع عن والتصرٌح

 'prod1 and prod2 have no key values. 
Dim prod1 = New With {.Name = "paperclips", .Price = 1.29} 
Dim prod2 = New With {.Name = "paperclips", .Price = 1.29} 
 
 'The following line displays False, because prod1 and prod2 have no 
 'key properties. 
Console.WriteLine(prod1.Equals(prod2(( 
 
 'The following statement displays True because prod1 is equal to itself. 
Console.WriteLine(prod1.Equals(prod1(( 
 

 كٌفٌة ٌوضح الذي التالً المثال فً كما متساوٌة المفتاحٌة خصائصهما قٌمة كانت إذا متساوٌٌن المجهول النوع لنفس متؽٌرٌن قٌمة وتكون
 المساواة هذه فحص

Dim prod3 = New With {Key .Name = "paperclips", Key .Price = 1.29} 
Dim prod4 = New With {Key .Name = "paperclips", Key .Price = 1.29} 
 'The following line displays True, because prod3 and prod4 are 

 'instances of the same anonymous type, and the values of their 
 'key properties are equal. 
Console.WriteLine(prod3.Equals(prod4(( 
 
Dim prod5 = New With {Key .Name = "paperclips", Key .Price = 1.29} 
Dim prod6 = New With {Key .Name = "paperclips", Key .Price = 1.29, OnHand = 423} 
 'The following line displays False, because prod5 and prod6 do not  
 'have the same properties. 
Console.WriteLine(prod5.Equals(prod6(( 
 
Dim prod7 = New With {Key .Name = "paperclips", Key .Price = 1.29, OnHand = 24} 
Dim prod8 = New With {Key .Name = "paperclips", Key .Price = 1.29, OnHand = 423} 
 'The following line displays True, because prod7 and prod8 are 
 'instances of the same anonymous type, and the values of their 
 'key properties are equal. The equality check does not compare the 
 'values of the non-key field. 

Console.WriteLine(prod7.Equals(prod8(( 
 

  فقط نهقراءة انقببهة انقيى

 أن حٌن فً فقط للقراءة قابلة Price و Name الحقول السابق المثال فً prod8 فً فمثلا المفتاحٌة الخصائص قٌم تؽٌٌر  لا ٌمكن

 قٌمته تؽٌٌر ٌمكن OnHamd الحقل

 'The following statement will not compile, because Name is a key 
 'property and its value cannot be changed. 
 'prod8.Name = "clamps" 

 'OnHand is not a Key property. Its value can be changed. 
prod8.OnHand = 22 

 Anonymous Types from Query Expressions الاستعلام تعابير من المجهولة الأنواع

 ٌحدث وهذا العمود بٌانات لٌحمل موجود نوع استخدام ٌمكنها الإمكان فعند مجهولة أنواع إنشاء دوما لا تتطلب الاستعلام  تعابٌر
 عن عبارة Customers ٌكون التالً المثال ففً سجل كل من واحد حقل أو البٌانات مصدر من كاملة سجلات إما الاستعلام ٌعٌد عندما

 الاستعلام نتائج فً الخصائص هذه من أكثر أو واحدة تضمٌن ٌمكنك بحٌث الخصائص من العدٌد تمتلك والفئة Customer فئات مجموعة

 ٌكون Custs1 فـ معروفة أنواع من عناصر ٌجلب الاستعلام لأن مجهول نوع لأي حاجة لا ٌوجد الأولٌٌن المثالٌن ففً ترٌده ترتٌب وبؤي

 فً عنصر كل لأن Customers الأؼراض من مجموعة هو Custs2 و String النوع من cust.Name لأن string النوع من

Customers ؼرض هو Customer الاستعلام بواسطة جلبه تم العنصر وكامل 

Dim custs1 = From cust In customers_  
             Select cust.Name 

 
Dim custs2 = From cust In customers_  
             Select cust 
 

 و ID و معٌن هدؾ أجل من Addresses و Names عن الاستعلام ٌمكنك حٌث متوفرة دائما لا تكون المسماة فالأنواع ذلك ومع

Numbers و Location أن بدون ترتٌب وبؤي الخصائص من تركٌبة أٌة اختٌار من المجهولة الأنواع تمكنك فهنا آخر هدؾ أجل من 

 الخصائص من تركٌبة لكل مجهول نوع بإنشاء المترجم ٌقوم ذلك عن وبدلا النتٌجة لٌحمل جدٌد مسمى نوع عن للتصرٌح البداٌة فً تضطر
 تلك من مجهول نوع بإنشاء ٌقوم ذلك أجل  ومنcustomers فً Customer ؼرض كل من ID و Name فقط ٌحدد التالً الاستعلام فمثلا

 الخصائص

Dim custs3 = From cust In customers_  
             Select cust.Name, cust.ID 
 

 خصائص وتكون Cust.Id و cust.Name الاستعلام بارامترات من أخذها ٌتم المجهول النوع لخصائص العائدٌن والنوع الاسم من وكل

 مجموعة هً النتٌجة تكون التالٌة For...Each حلقة فً cust3 تنفٌذ وعند دوما مفتاحٌة خصائص الاستعلام ٌنشئها التً المجهول النوع

 ID و Name مفتاحٌتٌن خاصٌتٌن مجهولة تمتلك أنواع

For Each selectedCust In custs3 
    Console.WriteLine(selectedCust.ID & ": " & selectedCust.Name) 
Next 

 انًجهىنة الأنىاع نسحخذو يحى جحذيذ

 ترٌد كنت إن فمثلا الأفضل هو الخٌار هذا كان إذا فٌما التفكٌر علٌك ٌجب مجهول نوع من بؤنه ؼرض عن بالتصرٌح تقوم أن  قبل
 النوع عندها ٌكون الكاملة الفئة تحتوٌها التً والطرائق الحقول بقٌة إلى بحاجة ولست المعلومات حقول بعض لٌحتوي مإقت ؼرض إنشاء

 تؽٌٌر ترٌد كنت إن أو تصرٌح كل عند الخصائص من مختلفة مجموعة انتقاء ترٌد عندما ملائمة المجهولة الأنواع وتكون جٌدا حلا المجهول
 عنهم التصرٌح عندها ٌمكنك ثابت بترتٌب الخصائص نفس تحمل أؼراض عدة على ٌحتوي مشروعك كان وإن الخصائص هذه ترتٌب
 ذلك وٌكون Product الفئة من متؽٌرات عدة تعرٌؾ ٌمكن ملائم بانً باستخدام فعندها فئة بانً باستخدام المسماة الأنواع باستخدام بسهولة

 النوع مجهولة متؽٌرات عدة استخدام من أسهل

 'Declaring instances of a named type. 
Dim firstProd1 As New Product("paperclips", 1.29( 
Dim secondProd1 As New Product("desklamp", 28.99( 
Dim thirdProd1 As New Product("stapler", 5.09( 
 'Declaring instances of an anonymous type. 
Dim firstProd2 = New With {Key .Name = "paperclips", Key .Price = 1.29} 
Dim secondProd2 = New With {Key .Name = "desklamp", Key .Price = 28.99} 
Dim thirdProd2 = New With {Key .Name = "stapler", Key .Price = 5.09} 
 

 بالأنواع ٌفترض السابق المثال ففً الخصائص أسماء فً الطباعٌة الأخطاء التقاط ٌمكنه المترجم أن فً المجهولة للأنواع أخرى فائدة وتكمن
firstProd2 و secondProd2 و thirdProd2 عن بالتصرٌح الخطؤ طرٌق عن قمت ذلك ومع المجهول النوع لنفس متؽٌرات تكون أن 

thirdProd2 عن مختلؾ نوع وهو اللاحقة الطرائق بؤحد firstProd2 و secondProd2 

 'Dim thirdProd2 = New With {Key .Nmae = "stapler", Key .Price = 5.09} 
 'Dim thirdProd2 = New With {Key .Name = "stapler", Key .Price = "5.09"} 

 'Dim thirdProd2 = New With {Key .Name = "stapler", .Price = 5.09} 

 و secondProd2 و firstProd2 أن فمع المعروفة الأنواع على لا تنطبق المجهولة الأنواع لاستخدام حدود هناك أنه هو الأهم والأمر

thirdProd2 ًالكود فً معروؾ كنوع ظهوره توقع ولا ٌمكن متوفر ؼٌر المشترك المجهول فالمتؽٌر المجهول النوع لنفس متؽٌرات ه 

 البٌانات لتبادل ملائم ؼٌر المجهول النوع بالنتٌجة فٌكون متؽٌر حقل عن للتصرٌح الطرٌقة توقٌع لتحدٌد المجهول النوع استخدام ٌمكن فمثلا
 الطرائق عبر

Lambda Expressions 

 

 بدون اسم تحتسب وتعٌد قٌمة وحٌدة كما ٌمكن استخدامها فً التعابٌر التً تطلب Function هو وظٌفة Lambda Expressionالـ 

  والمثال التالً عن هذه التعابٌر ٌؤخذ قٌمة وٌعٌد الناتج بعد إضافة واحد لهاDelegateإجراءات مفوضة 

Function (num As Integer) num + 1 

 كما ٌمكنك إسناد هذه الوظٌفة لمتؽٌر وتمرٌر القٌمة له

Dim add1 = Function(num As Integer) num + 1 

Console.WriteLine(add1(5)) 

 كما ٌمكنك تعرٌؾ وتنفٌذ الوظٌفة بنفس الوقت

Console.WriteLine((Function(num As Integer) num + 1)(5)) 

 كقٌمة معادة عند استدعاء وظٌفة أو تمرٌرها كوسٌط لإجراء مفوض ففً المثال التالً Lambda Expressionsكما ٌمكن أن تستخدم 

 Integer حٌث تطبق الطرٌقة فحص بولٌانً لوسٌط من النوع testResult بولٌانٌة كوسائط للإجراء Lambda Expressionsتستخدم 

 False إن كانت قٌمته Failure أو True هً Lambda Expression إذا كانت قٌمة Successوٌظهر القٌمة 

Module Module2 

 

    Sub Main() 

        ' The following line will print Success, because 4 is even. 

        testResult(4, Function(num) num Mod 2 = 0) 

        ' The following line will print Failure, because 5 is not > 10. 

        testResult(5, Function(num) num > 10) 

    End Sub 

 

    ' Sub testResult takes two arguments, an integer value and a  

    ' Boolean function.  

    ' If the function returns True for the integer argument, Success 

    ' is displayed. 

    ' If the function returns False for the integer argument, Failure 

    ' is displayed. 

    Sub testResult(ByVal value As Integer, ByVal fun As Func(Of Integer, 

Boolean)) 

        If fun(value) Then 

            Console.WriteLine("Success") 

        Else 

            Console.WriteLine("Failure") 

        End If 

    End Sub 

 

End Module 

 بإنشاء Compiler حٌث ٌقوم المترجم Linq هً الأساس لكثٌر من معاملات الاستعلام Lambda Expressions تكون تعابٌر 

 Take و By و Order و Select و Where للقٌام بالعملٌات الحسابٌة للطرائق الخاصة بالاستعلام مثل Lambda Expressionsتعابٌر 

Whileًفعلى سبٌل المثال انظر الاستعلام التال  

Dim londonCusts = From cust In db.Customers  

                  Where cust.City = "London"  

                  Select cust 

 حٌث ستتم ترجمته إلى الكود التالً

Dim londonCusts = db.Customers _ 

    .Where(Function(cust) cust.City = "London") _ 

    .Select(Function(cust) cust) 

 

 وتكون صٌؽتها على الشكل

 هذه التعابٌر لا تملك اسما 

  لا ٌمكن استخدام المعدلات معها مثلOverloads أو Overrides 

  لا تستخدم قسمAS لتحدٌد نوع القٌمة المعادة وبدلا عن ذلك ٌكون نوع القٌمة المعادة هو نوع القٌمة التً ٌشكلها جسم الإجراء فإن 

  فتكون القٌمة المعادة بولٌانٌة "Cust.City = "Londonكان جسم الإجراء مثلا 

  جسم الإجراء ٌجب أن ٌكون تعبٌر ولٌس تصرٌح وٌمكن أن ٌحتوي على استدعاء لوظٌفةFunction ًولكنه لا ٌمكن أن ٌستدع 

 Subإجراء 

  لا ٌوجد تعبٌرReturnوتكون القٌمة المعادة هً قٌمة ذلك التعبٌر الذي ٌشكل جسم الوظٌفة  

  لا ٌوجد تعبٌرEnd 

 ٌجب أن تكون جمٌع الوسائط محددة النوع أو تكون جمٌعها بؤنواع بالإشارة 

 ؼٌر مسموح بالوسائط الاختٌارٌة 

  الوسائطGenericؼٌر مسموح بها  

  سٌكون بسٌطا وؼٌر معقدLambda Expressionونتٌجة لهذه القواعد سنرى أن أي تعبٌر 

 بؤنها محددة ولها جمٌع حقوق الوصول كؤي كود مكتوب فً الطرٌقة التً تحتوٌها Methods مع الوظائؾ Lambda Expressionتشترك 

 Lambdaوهذا ٌتضمن الوصول إلى متؽٌرات الأعضاء والوظائؾ وجمٌع المؽٌرات الموجودة فً الوظٌفة التً تحتوي التعبٌر 

Expression ففً المثال التالً المتؽٌر target هو محلً بالنسبة لـ makeTheGame والطرٌقة التً تم تحدٌد التعبٌر Lambda 

Expression ًفٌها ه playTheGame لاحظ أن القٌمة المعادة من التعبٌر Lambda Expressionتعٌٌنها لـ   ٌتم takeAGuess ًف 

Main ًمازالت تستطٌع الوصول للمتؽٌر المحل target 

Module Module1 

 

    Sub Main() 

       ' Variable takeAGuess is a Boolean function. It stores the target 

       ' number that is set in makeTheGame. 

       Dim takeAGuess As gameDelegate = makeTheGame() 

 

       ' Set up the loop to play the game. 

       Dim guess As Integer 

       Dim gameOver = False 

       While Not gameOver 

         guess = CInt(InputBox("Enter a number between 1 and 10 (0 to quit)", "Guessing Game", "0")) 

            ' A guess of 0 means you want to give up. 

            If guess = 0 Then 

                gameOver = True 

            Else 

                ' Tests your guess and announces whether you are correct. Method takeAGuess 

                ' is called multiple times with different guesses. The target value is not  

                ' accessible from Main and is not passed in. 

                gameOver = takeAGuess(guess) 

                Console.WriteLine("Guess of " & guess & " is " & gameOver) 

            End If 

        End While 

 

    End Sub 

 

    Delegate Function gameDelegate(ByVal aGuess As Integer) As Boolean 

 

    Public Function makeTheGame() As gameDelegate 

 

        ' Generate the target number, between 1 and 10. Notice that  

        ' target is a local variable. After you return from makeTheGame, 

        ' it is not directly accessible. 

        Randomize() 

        Dim target As Integer = CInt(Int(10 * Rnd() + 1)) 

 

        ' Print the answer if you want to be sure the game is not cheating 

        ' by changing the target at each guess. 

        Console.WriteLine("(Peeking at the answer) The target is " & target) 

 

        ' The game is returned as a lambda expression. The lambda expression 

        ' carries with it the environment in which it was created. This  

        ' environment includes the target number. Note that only the current 

        ' guess is a parameter to the returned lambda expression, not the target.  

 

        ' Does the guess equal the target? 

        Dim playTheGame = Function(guess As Integer) guess = target 

        Return playTheGame 

 

    End Function 

 

End Module 

 

 Lambda فعندما ٌتم تنفٌذ التعبٌر Lambda Expressionوٌستعرض المثال التالً مجالا عرٌضا من حقوق الوصول المعششة فً 

Expression من Main كـ aDel حقل فً الفئة  ) ٌستخدم العناصر التالٌةaField – خاصٌة فً الفئة aProp – وسٌط للإجرائٌة 

functionWithNestedLambda هو level1 – متؽٌر محلً لـ functionWithNestedLambda هو localVar – وسٌط للتعبٌر 

Lambda Expression المعشش هو level2)  

Module Module3 

 

    Sub Main() 

        ' Create an instance of the class, with 1 as the value of  

        ' the property. 

        Dim lambdaScopeDemoInstance = New LambdaScopeDemoClass _ 

            With {.Prop = 1} 

 

        ' Variable aDel will be bound to the nested lambda expression   

        ' returned by the call to functionWithNestedLambda. 

        ' The value 2 is sent in for parameter level1. 

        Dim aDel As aDelegate = _ 

            lambdaScopeDemoInstance.functionWithNestedLambda(2) 

 

        ' Now the returned lambda expression is called, with 4 as the  

        ' value of parameter level3. 

        Console.WriteLine("First value returned by aDel:   " & aDel(4)) 

 

        ' Change a few values to verify that the lambda expression has  

        ' access to the variables, not just their original values. 

        lambdaScopeDemoInstance.aField = 20 

        lambdaScopeDemoInstance.Prop = 30 

        Console.WriteLine("Second value returned by aDel: " & aDel(40)) 

    End Sub 

 

    Delegate Function aDelegate(ByVal delParameter As Integer) _ 

        As Integer 

 

    Public Class LambdaScopeDemoClass 

        Public aField As Integer = 6 

        Dim aProp As Integer 

 

        Property Prop() As Integer 

            Get 

                Return aProp 

            End Get 

            Set(ByVal value As Integer) 

                aProp = value 

            End Set 

        End Property 

 

        Public Function functionWithNestedLambda _ 

           (ByVal level1 As Integer) As aDelegate 

            Dim localVar As Integer = 5 

 

            ' When the nested lambda expression is executed the first  

            ' time, as aDel from Main, the variables have these values: 

            ' level1 = 2 

            ' level2 = 3, after aLambda is called in the Return statement 

            ' level3 = 4, after aDel is called in Main 

            ' locarVar = 5 

            ' aField = 6 

            ' aProp = 1 

            ' The second time it is executed, two values have changed: 

            ' aField = 20 

            ' aProp = 30 

            ' level3 = 40 

            Dim aLambda = Function(level2 As Integer) _ 

                              Function(level3 As Integer) _ 

                                  level1 + level2 + level3 + localVar _ 

                                  + aField + aProp 

 

            ' The function returns the nested lambda, with 3 as the  

            ' value of parameter level2. 

            Return aLambda(3) 

        End Function 

 

    End Class 

End Module 

 

 لإجراء مفوض Lambda Expression لتتوافق مع الإجراءات المفوضة فعندما تعٌن Lambda Expressionsكما ٌمكن تحوٌل 

Delegate ًٌمكنك تحدٌد أسماء الوسائط ولكن مع إؼفال أنواع البٌانات الخاصة بها تاركا مهمة تحدٌدها للإجراء المفوض ففً المثال التال 

 integer الذي هو عبارة عن إجراء مفوض ٌؤخذ وسٌطتٌن ExampleDel من النوع del لمتؽٌر اسمه Lambda Expressionٌتم تعٌٌن 

 integer ٌتطلب وسٌطا من النوع del لم ٌتم تحدٌدها ومع ذلك فـ Lambda Expression لاحظ أن أنواع المتؽٌرات فً stringو 

 ExampleDel كما تم تحدٌده عند تعرٌؾ stringووسٌطا آخر من النوع 

' Definition of function delegate ExampleDel. 

Delegate Function ExampleDel(ByVal arg1 As Integer, _ 

                             ByVal arg2 As String) As Integer 

 

' Declaration of del as an instance of ExampleDel, with no data  

' type specified for the parameters, m and s. 

Dim del As ExampleDel = Function(m, s) m 

 

' Valid call to del, sending in an integer and a string. 

Console.WriteLine(del(7, "up")) 

 

' Neither of these calls is valid. Function del requires an integer 

' argument and a string argument. 

'Console.WriteLine(del(7, 3)) 

'Console.WriteLine(del("abc")) 

 

 Nothing إذا كان القٌمة False إذا كان الوسٌط ٌمتلك قسمة أو True لٌعٌد القٌمة Lambda Expressionفً المثال التالً ٌتم تحدٌد 

Dim notNothing = Function(num? As Integer) _ 

             num IsNot Nothing 

Dim arg As Integer = 14 

Console.WriteLine("Does the argument have an assigned value?") 

Console.WriteLine(notNothing(arg)) 

 

  العنصر الأخٌر فً مصفوفةIndex ٌعٌد Lambda Expressionوالمثال التالً ٌحدد 

Dim numbers() As Integer = {0, 1, 2, 3, 4, 5, 6, 7, 8, 9} 

Dim lastIndex = Function(intArray() As Integer) _ 

                    intArray.Length - 1 

For i = 0 To lastIndex(numbers) 

    numbers(i) = numbers(i) + 1 

Next 

 Lambda Expressionsجعببير نًذا في انعًق 

 

 حٌث ٌمكنك إعادتها كقٌمة من وظٌفة أو تمرٌرها كمحدد لوظٌفة 2008تعتبر تعابٌر لمدا من الإضافات المفٌدة فً فٌجول باٌزٌك 

 التً تضٌؾ إمكانٌات استعلامٌة قوٌة لبرمجة البٌانات فً Linqأخرى حٌث تمت إضافتها للؽة الباٌزٌك كدعم للؽة الاستعلامات المضمنة 

 فٌجول باٌزٌك وعندما تبدأ باستخدام تعابٌر لمدا سترى القوة والمرونة الكامنة فٌها

 يب هي جعببير نًذا

 بحٌث ٌؤخذ قٌمة Integer وٌعٌد Integer ٌشكل الكود التالً مثالا عن تعرٌؾ تعبٌر لمدا أساسً فهو ٌعرؾ تعبٌر لمدا ٌؤخذ 

 2الدخل وٌعٌدها مضروبة بـ 

Dim doubleIt as Func(Of Integer, integer) = _ 
        Function(x As Integer) x * 2 

 ٌعٌد نوعا ٌحدده المحدد الأخٌر Delegate وهو فً الأساس إجراء مفوض 2008 من الأنواع الجدٌدة فً فٌجول باٌزٌك Funcوالنوع 

 الأمر الذي ٌمكنك من System.Core.dll معرؾ فً المجمع Funcوٌمكنك من تمرٌر أربعة محددات تسبق ذلك المحدد والنوع المفوض 

 مختلفة Overloadsالاستفادة منه فورٌا وذلك لأن المجمع المذكور ٌتم استٌراده تلقائٌا عندما ننشئ تطبٌقا جدٌدا وٌمثل الكود التالً تحمٌلات 

 Funcلـ 

Dim f0 As Func(Of Boolean) 
Dim f1 As Func(Of Integer, Boolean) 
Dim f4 As Func(Of Integer, Integer, Integer, Integer, Boolean) 

 مفوض ٌؤخذ أربعة محددات من f4 و Boolean وٌعٌد Integer مفوض ٌؤخذ f1 و Boolean هو مفوض ٌعٌد قٌمة f0ففً المثال السابق 

 وتكمن النقطة الأساسٌة فً التعبٌر لمدا هو أنه مفوض قابل للاستدعاء تماما كالمفوضات فً Boolean وٌعٌد قٌمة من النوع Integerالنوع 

 فمن ناحٌة المساواة الٌمنى فً قطعة الكود الأولى ٌمكنك رإٌة الصٌؽة الجدٌدة للتعابٌر لمدا فهً تبدأ بالكلمة المحجوزة 2005فٌجول باٌزٌك 

Function متبوعة بقائمة من المحددات وتعبٌر وحٌد ففً المثال السابق ٌؤخذ تعبٌر لمدا محددا واحدا من النوع Integer ونلاحظ عدم 

 تلقائٌا وفً هذه الحالة بما أن Return وذلك لأن المترجم ٌعرؾ النوع المإسس علٌه التعبٌر وبهذا ٌقوم بتمرٌر عبارة Returnوجود تعبٌر 

x هو من النوع Integer ًونتٌجة المعادلة ه Integer ًلهذا فنتٌجة تعبٌر لمدا ه Integer أٌضا وٌمكن السحر فً تعابٌر لمدا فً أنه 

 ٌمكن استخدامها كمفوض بسٌط كما نرى فً المثال

Dim doubleIt As Func(Of Integer, Integer) = _ 
        Function(x As Integer) x * 2 
Dim z = doubleIt(20) 

 . ٌمرر لهInteger وأنت بهذا قمت بإنشاء تعبٌر لمدا ٌضاعؾ قٌمة أي 40 هً zفإن نفذت الكود السابق سترى أن القٌمة المخزنة فً 

 دعنا الآن نتفحص مثال معقد أكثر باستخدام تعابٌر لمدا

Dim mult As Func(Of Integer, Func(Of Integer, Integer)) = _ 
        Function(x As Integer) Function(y As Integer) x * y 

 Integer وٌعٌد تعبٌر لمدا كقٌمة له والذي أخذ بدوره قٌمة Integer تعبٌر لمدا معقد قلٌلا فهو ٌؤخذ كدخل له محدد من النوع multوٌعتبر 

  كما ٌمكننا إعادة تقسٌم التعبٌر السابق على أسطر من أجل توضٌح الكودIntegerوٌعٌد قٌمة 

Dim mult As Func(Of Integer, Func(Of Integer, Integer)) = _ 
        Function(x As Integer)  _ 
                Function(y As Integer) x * y 

فتعبٌر لمدا الخارجً ٌحتوي تعبٌر لمدا آخر الذي ٌستخدم من قبل المترجم كقٌمة معادة وٌكون التوقٌع الخاص بتعبٌر لمدا الداخلً مماثلا 

 فً القٌمة المعادة من تعبٌر لمدا الخارجً حٌث ٌقوم المترجم بترجمة التعبٌر بؤكمله دون Func(Of Integer, Integer)لتوقٌع المفوض 

 مشاكل وٌمكننا رإٌة تعبٌر لمدا هذا كما ٌلً

Dim mult_10 = mult(10) 
Dim r = mult_10(4) 

 mult_10 والنوع المعاد من 10 ٌعٌد تعبٌر لمدا ٌؤخذ محدد وٌضربه بـ Mult(10) وبما أن mult(10) كـ mult_10فالسطر الأول ٌحدد 

 وٌكون 40 هً r بهذا ستكون القٌمة المخزنة فً 4 ممررا له القٌمة mult_10 والسطر الثانً ٌستدعً Func(Of Integer, Integer)هو 

 مصنع لتعابٌر لمدا فهو ٌعٌد تعبٌر لمدا مخصص بالمحدد الأول وستلاحظ أن تعبٌر لمدا الداخلً ٌستخدم mult وٌعتبر Integer هو rنوع 

 محدد تعبٌر لمدا الخارجً ولكن فترة حٌاة تعبٌر لمدا الداخلً تتجاوز فترة حٌاة تعبٌر لمدا الخارجً

 جعببير نًذا كبسحذعبءات

لاحظ الإجراء  التالً الذي .  بما أن تعابٌر لمدا هً ببساطة مفوضات لذا ٌمكنك استخدامها فً أي مكان ٌمكن استخدام المفوض فٌه

 ٌؤخذ مفوض كمحدد له وٌستدعً مفوض من أجل كل عنصر فً القائمة

Delegate Function ShouldProcess(Of T) (element As T) As Boolean 
 
Sub ProcessList(Of T) ( _ 
                        Elements As List(Of T), shouldProcess As ShouldProcess(Of T)) 
 
        For Each elem in elements  
                If shouldProcess(elem) Then 
                        ' Do some processing here 
                End If 
        Next 
End Sub 

 ستمر على كل عنصر من القائمة وتتحقق فٌما إذا كان علٌها ProcessListوٌكون المثال التالً تطبٌقا قٌاسٌا على المفوضات فالطرٌقة 

 علٌك إنشاء وظٌفة تمتلك نفس توقٌع 2005معالجة العنصر ثم تقوم ببعض المعالجة وحتى تتمكن من استخدام هذا فً فٌجول باٌزٌك 

 ProcessListالمفوض ثم تمرر عنوان تلك الوظٌفة إلى الإجراء 

Class Person 
        Public age As Integer 
End Class 
 
Function _PrivateShouldProcess(person As Person) As Boolean 
        Return person.age > 50 
End Function 
 
Sub DoIt() 
        Dim list As New List(Of Person) 
        'Obtain list of Person from a database, for example 
        ProcessList(list, AddressOf _PrivateShouldProcess) 
End Sub 

وهذا ٌسبب بعض الإزعاج فؽالبا علٌك البحث فً توثٌق الكود لمعرفة ماذا ٌمثل توقٌع المفوض ثم ٌجب علٌك مطابقته كلٌا وإن احتجت 

 . مع عدة إجراءات ستقوم بإنشاء العدٌد من الوظائؾ الخاصةProcessListلاستدعاء 

 دعنا نرى الآن كٌؾ ٌمكننا استدعاء هذا الإجراء باستخدام تعابٌر لمدا

Class Person 
        Public age As Integer 
End Class 
 
Sub DoIt() 
        Dim list As new List(Of Person) 
        'Obtain list of Person from a database, for example 

        ProcessList(list, Function(person As Person) person.age > 50) 
End Sub 

فباستخدام تعابٌر لمدا لم ٌعد هناك حاجة لإنشاء وظٌفة خاصة للقٌام بمنطق المعالجة حٌث ٌتم تعرٌؾ المفوض فً النقطة التً سٌستخدم فٌها 

وهذا أفضل من تعرٌفه ضمن وظٌفة خاصة فً مكان ما وفقدان محلٌتها باستخدام الطرٌقة الخاصة وبهذا أنت ترى قوة تعابٌر لمدا وتسهٌلها 

 لعملٌة قراءة وصٌانة الكود الخاص بك

 نًبرا جى جقذيى جعببير نًذا

 كان ٌجب إضافة مجموعة من الإمكانٌات الجدٌدة للؽة فٌجول باٌزٌك ومن ضمنها كانت تعابٌر Linq من أجل دعم استعلامات لٌنك 

 افترض أنه لدٌنا الاستعلام التالً. لمدا

Dim q = From p In Process.GetProgesses() _ 
               Where p.PriorityClass = ProcessPriorityClass.High _ 
               Select P 

 Process.GetProcessesفلكً ٌتم ترجمة هذا التعبٌر ٌجري الكثٌر من العمل تحت الؽطاء فالمترجم سٌقوم بالمرور عبر المجموعة 

 علٌها وٌعٌد قائمة بالعملٌات التً تطابق ذلك الشرط كما نلاحظ وجود تعبٌر فٌجول باٌزٌك داخل Whereوٌطبق المرشح الموجود فً قسم 

 وذلك لتطبٌق المرشح وهنا ٌقوم المترجم بإنشاء تعبٌر لمدا من أجل p.PriorityClass = ProcessPriorityClass.High هو Whereقسم 

  وٌطبقه على كل عنصر فً قائمة العملٌاتWhereالمرشح الموجود فً قسم 

Dim q = Process.GetProcesses().Where( _ 
        Function(p) p.PriorityClass = ProcessPriorityClass.High) 

وأساسا ٌشكل التعبٌر لمدا اختصارا للمترجم من أجل اختصار عملٌة إنشاء الطرق وربطها مع المفوضات حٌث ٌقوم بكل ذلك من أجلك 

والفائدة التً نجنٌها من تعابٌر لمدا ولا نجنٌها عند استخدام الوظائؾ والمفوضات هً أن المترجم هنا ٌستخدم الاستدلال المحلً على النوع 

 تعبٌر لمدا Where بناء على الاستخدام وفً هذه الحالة ٌحدد التعبٌر فً قسم pعلى تعابٌر لمدا ففً المثال السابق ٌتم تحدٌد نوع المحدد 

وٌقوم المترجم بالاستدلال آلٌا على نوع القٌمة المعادة من التعبٌر لمدا بحٌث تعتبر مٌزة الاستدلال المحلً على النوع المدعومة من قبل 

 المعالج من الإضافات القوٌة لفٌجول باٌزٌك

 الاسحذلال انًحهي عهى اننىع

 تقدٌم مٌزة الاستدلال المحلً على النوع القوٌة ٌعنً أنه لم ٌعد علٌك أن تقلق حول تحدٌد النوع الملائم لكل متؽٌر وبالتالً فهً 

تمكنك من القٌام بالعدٌد من الأمور التً كانت تبدو مستحٌلة فالاستدلال على النوع المعاد من تعابٌر لمدا مفٌد جدا فإن كان لدٌك نوع مفوض 

 ترٌد ربطه مع تعبٌر لمدا لم ٌعد علٌك تحدٌد نوع جمٌع المحددات

Dim lambda As Func(Of Integer, Integer) = Function(x) x * x 

 وٌعٌد محدد من النوع Integer وهو مفوض ٌؤخذ محدد من النوع Func(Of Integer, Integer)ففً هذا المثال ٌكون نوع تعبٌر لمدا هو 

Integer وكنتٌجة لهذا فالمترجم ٌستدل آلٌا على أن المحدد x العائد لتعبٌر لمدا هو من النوع Integer ًوالقٌمة المعادة من التعبٌر لمدا ه 

Integerًأٌضا كما ٌمكنك الاستفادة من الاستدلال على نوع تعابٌر لمدا عندما تستدعً طرٌقة تؤخذ مفوضا لاحظ الكود التال  

Delegate Function ShouldProcess(Of T) (element As T) As Boolean 
 
Sun ProcessList(Of T) (_ 
                        Elements As List(Of T), shouldProcess As ShouldProcess(Of T)) 
        ' Method body removed for brevity 
End Sub 

  تعبٌر لمدا وٌمكن استدعاإها على الشكلProcessListفً هذه الحالة تؤخذ الوظٌفة 

Sub DoIt() 
        Dim list As new List(Of A) 
        ' fill or obtain elements in list 
        ProcessList(list, Function(a) a.x > 50) 

End Sub 

 .لاحظ أننا لم نحدد نوع المحدد الممرر للتعبٌر لمدا كما فعلنا سابقا وذلك لأن المعالج ٌستدل علٌه بنفسه

 ProcessListكٌؾ ٌمكن حدوث شئ كهذا؟ فً الحقٌقة هناك عدة مستوٌات من الاستدلال على النوع فً هذا المثال ففً البداٌة ٌرى المترجم 

 هً المحدد الأول list وٌرى المترجم أن ProcessList فً استدعاء ShouldProcess(Of T) كدخل له و list(Of T)كإجراء عادي ٌؤخذ 

 وٌستدل من هذا Person من النوع T فٌقرر المترجم أن T وبما أن المحدد الثانً لا ٌوفر تلمٌحات حول ماهٌة نوع list(Of Person)وأنها 

 ShouldProcess(Of T) وبهذا ٌستدل على أن المحدد الثانً هو من النوع Person هو من النوع ShouldProcess(Of T)على أن محدد 

 وأخٌرا بما أن تعبٌر لمدا لا ٌقدم نوع المحدد الخاص به والمترجم ٌعرؾ أن نوع المحدد ٌعتمد على توقٌع 

 وٌعتبر هذا نوعا قوٌا من الاستدلال على النوع فلٌس Person هو a وقد استدل على أن نوع المحدد ShouldProcess(Of T)المفوض 

علٌك معرفة نوع محددات المفوض عندما تبنً تعبٌر لمدا وفً الحقٌقة من الأفضل ترك المترجم ٌقوم بذلك العمل نٌابة عنك والاستدلال على 

نوع النتٌجة بهذه الطرٌقة مفٌد حقٌقة إن لم ٌكن لدٌك نوع مفوض وترٌد من المترجم أن ٌقوم بتصنٌعه من أجلك علما بؤن هذه المٌزة متوفرة 

 فً فٌجول باٌزٌك فقط

Dim lambda = Function(x As Integer) x * x 

 أٌضا كنتٌجة Integer فالمترجم ٌستدل آلٌا على أن القٌمة المعادة هً من النوع Integer هو من النوع xففً المثال السابق بما أن المحدد 

المعادلة الموجودة فً التعبٌر وبما أن تعبٌر لمدا لا ٌمتلك نوعا لهذا ٌقوم المترجم بتصنٌع مفوض مجهول ٌطابق شكل تعبٌر لمدا وٌربط ذلك 

وهذه مٌزة عظٌمة لأنها تعنً أنه ٌمكنك إنشاء تعابٌر لمدا بسرعة بدون أن تحتاج لتعرٌؾ الأنواع المفوضة . النوع المفوض بتعبٌر لمدا

فكم مرة كنت فً وضع تحتاج فٌه لتطبٌق مجموعة من المتؽٌرات وتحتاج إلى فعل ذلك فً العدٌد من الأماكن ففً الكود التالً . الخاصة بها

 مرت عدة حالات مشابهة وعادة ٌمكننا معالجة ذلك بحٌث ٌمكن التحقق من الشرط فً مكان واحد بدلا من التشتت فً أرجاء الوظٌفة

Class Motorcycle 
        Public color As String 
        Public CC As Integer 
        Public weight As Integer 
End Class 
 
Sub PrintReport(motorcycle As New Motorcycle) 
        If motorcycle.color = "Red" And motorcycle.CC = 600 And _ 
                Motorcycle.weight > 300 And Motorcycle.weight < 400 Then 
 
                ' do something here 
        End If 
 
        ' do something here 
 
        If motorcycle.color = "Red" And motorcycle.CC = 600 And _ 
                Motorcycle.weight > 300 And Motorcycle.weight < 400 Then 
 
                ' do something here 
        End If 
End Sub 

وفً بعض الأحٌان ٌستخدم هذا التحقق فً هذه الوظٌفة فقط وٌمكننا إضافة إجراء فً الفئة لدعم تلك الوظٌفة فقط والقٌام بذلك ٌإثر على 

عملٌة صٌانة الكود فماذا لو قام أحد ما باستدعاء هذه الوظٌفة فً مكان آخر واحتجت للقٌام بتعدٌل ما وقد ٌإدي هذا فً بعض الفئات إلى 

 اقٌل فائدة لوجود العدٌد من المدخلات الإضافٌة فٌها إضافة إلى خرق منطق IntelliSenseوجود وظائؾ ٌصعب تعقبها جاعلا خاصٌة 

المحلٌة وإن قمنا بذلك باستخدام طرٌقة منفصلة مختلفة عندها ٌفضل أن تكون قرٌبة من الطرٌقة التً تستخدمها ومع وجود العدٌد من 

الأشخاص ٌعملون على نفس المشروع ٌصبح من الصعب صٌانة المحلٌة على المدى الطوٌل وهنا ٌؤتً استخدام تعابٌر لمدا وترك المترجم 

 ٌقوم آلٌا بإنشاء المفوضات وٌقوم باستخدامها عند الحاجة

Sub PrintReport(motorcycle As New Motorcycle) 
        Dim check = Function(m As Motorcycle) m.color =  "Red" And _ 

                                                                                                     m.CC = 600 And _ 
                                                                                                     m.weight > 300 And _ 
                                                                                                     m.weight < 400 
 
        If check(motorcycle) Then 
                ' do something here 
        End If 

 
        ' do something here 

 
        If check(motorcycle) Then 
                ' do something here 
        End If 
End Sub 

 لٌستخدم تعابٌر لمدا عوضا عن سٌئات الطرائق الخاصة حٌث سٌقوم المترجم Motorcycleقمنا هنا بتعدٌل منطق تفحص بعض شروط 

تلقائٌا بإنشاء النوع المفوض وٌقوم بالعمل لكً نستطٌع استدعاء تعابٌر لمدا أٌنما احتاج ذلك وهذه الطرٌقة مفٌدة لأنها تضع المنطق قرٌب من 

التصرٌح حٌث نقوم بتصنٌع نسخة واحدة وٌقوم المترجم بعدها بمعظم عملٌات الصٌانة وٌعتبر هذا مفٌدا لأنه ٌمكنك من بناء تعبٌر معقد 

 كجسم لتعبٌر لمدا وباستخدام الربط المتؤخر والاستدلال على النوع فً هذا السٌنارٌو فلا نحدد نوع تعبٌر لمدا أو المتؽٌر

Dim lambda = Function(x) x * x 

 وهذا ٌعنً أنه قد تم تفعٌل الربط المتؤخر System.Objectوهنا أٌضا ٌولد المعالج مفوض مجهول من أجلك ولكن ٌحدد نوع تعبٌر لمدا كـ 

 وٌعتبر هذا السٌنارٌو جٌدا بالنسبة لأولئك الذٌن ٌعتمدون على الربط Off على الوضع Option Strictفً هذا السٌنارٌو عندما ٌكون الخٌار 

معرؾ على الأنواع الممررة إلى * المتؤخر حٌث أن تعابٌر لمدا تدعم عملٌات الربط المتؤخر بشكل كامل ففً المثال السابق طالما أن المعامل 

 تعبٌر لمدا فسوؾ ٌعمل

Dim a = lambda(10) 
Dim b = lambda(CDec(10)) 
Dim c = lambda("This will throw an exception because " & _ 
                             "strings don't support the * operator") 

موجود فً مكتبات زمن التشؽٌل بالنسبة للنوع الممرر فسوؾ ٌجري كل شئ بشكل جٌد كما * وكما ترى من المثال السابق طالما أن المعامل 

 .أن تعابٌر لمدا تتؤقلم بشكل رائع مع الربط المتؤخر فً فٌجول باٌزٌك

 انكىد انًىنذ جحث انغطبء

 انظر للكود السابق.  بعدما استكشفنا تعابٌر لمدا دعنا نلقً نظرة على الكود الذي ٌتم تولٌده من قبل المترجم

Sub TestLambda() 
        Dim doubleIt As Func(Of Integer, Integer) = _ 
                        Function(x As Integer) X * 2 
        Console.WriteLine(doubleIt(10)) 
End Sub 

 هو مفوض والمفوضات هً مإشرات للوظائؾ فكٌؾ ٌقوم المترجم إذا بالعمل؟ فً هذه الحالة ٌقوم المترجم بإصدار Funcأنت تعلم أن 

 وظٌفة جدٌدة وٌربطها بمفوض ٌشٌر إلى تلك الوظٌفة الجدٌدة

Private Function $GeneratedFunction$(x As Integer) As Integer 
        Return x * 2 
End Function 
 
Sub TestLambda() 
        Dim doubleIt As Func(Of Integer, Integer) = _ 
                AddressOf $GeneratedFunction$ 

        Console.WriteLine(doubleIt(10) 
End Sub 

حٌث ٌؤخذ المترجم تعبٌر لمدا وٌنشئ وظٌفة جدٌدة بمحتوٌاته وٌؽٌر عبارة التصرٌح بحٌث ٌؤخذ تعبٌر لمدا عنوان الوظٌفة الجدٌدة المولدة 

 C معرؾ فً الفئة TestLambdaففً هذه الحالة ٌتم تولٌد الوظٌفة بنفس الأب الذي ٌحتوي على الطرٌقة التً تستخدم تعبٌر لمدا فإن كان 

 أٌضا ونلاحظ أن هذه الوظٌفة ؼٌر قابلة للاستدعاء وٌتم التصرٌح عنها باستخدام محدد Cفسوؾ ٌتم تعرٌؾ الوظٌفة الجدٌدة فً الفئة 

 Privateالوصول 

 جعببير نًذا ورفع انًحغيرات

 فً الأمثلة السابقة ٌشٌر جسم تعابٌر لمدا إلى متؽٌرات ٌتم تمرٌرها إلى تلك المتؽٌرات ومع ذلك تؤتً قوة تعابٌر لمدا مع ثمار رفع 

وتعبٌر لمدا ٌمكن أن ٌستخدم متؽٌرات مرتبطة أو متؽٌرات حرة لم ٌتم تعرٌفها ضمن . المتؽٌرات وجمٌع تعابٌر لمدا مبنٌة على مبدأ متشابه

التوقٌع الخاص بتعبٌر لمدا فالمتؽٌرات الحرة ممكن أن ٌكون قد تم التصرٌح عنها فً الإجراء المستدعً للتعبٌر فقد تكون متؽٌرات محلٌة أو 

محددات ممررة لذلك الإجراء والتعابٌر المرتبطة تكون تلك التً تم التصرٌح عنها فً جسم التعبٌر أو عناصر فً الفئة المحتوٌة للتعبٌر لمدا 

وهذا هام من أجل التمٌٌز بٌن المتؽٌرات المرتبطة والحرة فً تعابٌر لمدا الخاصة بك لأنها تإثر على دلالة . متضمنا الفئة الأب لتلك الفئة

تعبٌر لمدا والكود الذي ٌتم تولٌده وبالتالً ٌإثر على صحة برنامجك وهذا مثال ٌحتوي على تعابٌر لمدا تستخدم متؽٌرات مرتبطة وأخرى 

 حرة

Function MakeLambda() As Func(Of Integer, Integer) 
        Dim y As Integer = 10 
        Dim addTen As Func(Of Integer, Integer) = Function(ByVal x) x + y 
        Return addTen 
End Function 
 
Sub UseLambda() 
        Dim addTen = MakeLambda() 
        Consloe.WriteLine(addTen(5)) 
End Sub 

 ولكن ٌمكن أن تسؤل نفسك كٌؾ ٌعمل هذا؟ تحدد UseLambda على نافذة الكونسول عندما ٌتم استدعاء 15فهذا الكود سٌقوم بطباعة 

 ولكن التعبٌر لمدا ٌتم إعادته كنوع معاد من الوظٌفة y كمتؽٌر محلً والتعبٌر لمدا ٌستخدم y المتؽٌر MakeLambdaالوظٌفة 

MakeLambda والوظٌفة UseLambda تحصل على التعبٌر لمدا من الوظٌفة MakeLambda وتنفذ التعبٌر لمدا وٌبدو الأمر كما لو أن 

 فعندما نحصل على التعبٌر لمدا MakeLambda تنتهً مع نهاٌة الطرٌقة yففترة حٌاة المتؽٌر .  قد تم تذكره من قبل التعبٌر لمداyالمتؽٌر 

 خارج المجال وٌجب إزالة المساحة التً تحجزها فً المكدس وبطرٌقة ما ٌعلق هذا MakeLambda فسوؾ تصبح MakeLambdaمن 

 بالمتؽٌر المرفوع وكما ترى y ففً هذه الحالة ٌدعى المتؽٌر Variable Liftingالمتؽٌر مع تعبٌر لمدا وهذا ما ٌعرؾ برفع المتؽٌر 

فالمتؽٌرات المرفوعة تعتبر مٌزة برمجٌة قوٌة فالمترجم ٌقوم بالكثٌر من العمل من أجل تمكٌنك من إمساك حالة المتؽٌر حٌث ٌحفظها خارج 

 بحٌث تكون Closureمجال فترة حٌاتها الطبٌعٌة فعندما ٌصادؾ المترجم تعابٌر لمدا تستخدم متؽٌرات حرة ٌقوم برفع المتؽٌر إلى فئة تدعى 

فترة حٌاة هذه الفئة تمتد إلى ما بعد فترة حٌاة المتؽٌرات الحرة المستضافة داخلها وٌقوم المترجم بإعادة كتابة الوصول إلى المتؽٌرات فً 

 Closureالطرق لٌتم الوصول إلى نسختها الموجودة فً الفئة 

 MakeLambdaدعنا نسٌر مرة أخرى عبر المثال 

Dim MakeLambda() As Func(Of Integer, Integer) 
        Dim y As Integer = 10 
        Dim addTen As Func(Of Integer, Integer) = Function(ByVal x) x + y 

Return addTen 
End Function 

 تعبٌر حر وٌقوم المترجم بالكشؾ عن ذلك وٌتابع بإنشاء الفئة y مرتبط بمحدد التعبٌر لمدا ولكن المتؽٌر xوكما قمنا بالتحلٌل سابقا فالمتؽٌر 

Closureالتً تلتقط المتؽٌرات الحرة كما فً تعرٌؾ تعبٌر لمدا  

Public Class _Closure$__1 
        Public y As Integer 

        Public Function _Lambda$__1(ByVal x As Integer) As Integer 
                Return x + Me.y 
        End Function 
End Class 

 وٌتم تحوٌل المتؽٌر الحر بعدها إلى متؽٌر مرتبط داخل الفئة Closure وٌخزنه فً الفئة y ٌلتقط المتؽٌر Closureٌمكنك رإٌة أن متؽٌر 

Closureًكما ٌقوم المترجم بإعادة كتابة الطرٌقة التً تحتوي على التعبٌر لمدا لتبدو كما ٌل  

Function MakeLambda() As Func(Of Integer, Integer) 
        Dim Closure As New _Closure$__1 
        Closure.y = 10 
        Return AddressOf Closure._Lambda$__1 
End Function 

 وٌضبط قٌمته Closure الذي تم رفعه ضمن المتؽٌر y وٌعٌد كتابة المتؽٌر Closureٌمكنك الآن رإٌة كٌؾ ٌقوم المترجم بإنشاء المتؽٌر 

 ومن الهام ملاحظة أن المترجم ٌقوم برفع المتؽٌرات الحرة فً تعابٌر لمدا فقط Closureوٌعٌد ببساطة عنوان تعبٌر لمدا المخزن ضمن الفئة 

 انظر للمثال التالً.  الذي ٌبقى موجودا طالما أن تعبٌر لمدا بقً موجوداClosureوٌتم التقاط حالة المتؽٌر فً 

Sub Test() 
        Dim y As Integer = 10 
        Dim Lambda As Func(Of Integer, Integer) = Function(ByVal x) x + y 
        y = 20 
        Console.WriteLine(Lambda(5)) 
End Sub 

 إذا؟ المترجم ٌقوم بالتقاط وإعادة كتابة جمٌع المتؽٌرات 25فلماذا .  فقد أصبت25ما هً القٌمة التً تظهر عند تنفٌذ الوظٌفة السابقة؟ إن قلت 

  كالتالClosureً إلى نسخة yالحرة 

Sub Test() 
        Dim Closure As New $Closure_Compiler_Generated_Name$ 
        Closure.y = 10 
        Dim Lambda = AddressOf Closure.Lambda_1 
        Closure.y = 20 
        Console.WriteLine(Lambda(5)) 
End Sub 

 وهذا هام جدا لأنه 5 + 20 وبهذا فعندما ٌتم تنفٌذ تعبٌر لمدا ٌعٌد 20 قد تؽٌرت إلى yففً الوقت الذي ٌتم تنفٌذ تعبٌر لمدا فٌه تكون قٌمة 

 انظر للمثال التالً.  وحٌد قد ترى تصرفات ؼرٌبةClosureعندما نؤتً للحدٌث عن الحلقات وأن المتؽٌرات الحرة ٌتم التقاطها فً 

Sub Test() 
        For I = 1 To 5 
                StartThread(Function() I + 10) 
        Next 
End Sub 

 فٌمكن أن تكون الحلقة قد Closure ٌنشئ مسارا جدٌدا وٌطبع النتٌجة على الكونسول وطالما أنه تم التقاطه إلى StartThreadافرض أن 

 فً الوقت الذي ٌقوم المسار فٌه باستدعاء تعبٌر لمدا وفً هذه الحالة فالبرنامج قد لا ٌطبع النتٌجة المتوقعة وبدلا عن ذلك علٌك Iؼٌرت قٌمة 

 رإٌة المتؽٌر الملتقط داخل الحلقة

Sub Test 
        For I = 1 To 5 
                Dim x = I 
                StartThread(Function() x + 10) 
        Next 

End Sub 

 والبرنامج سٌطبع القٌم كما هو متوقع ومن الهام جدا معرفة أٌة متؽٌرات سٌتم رفعها عندما سٌتم Closure فً xفالكود سٌلتقط الآن قٌمة 

 .تنفٌذ تعبٌر لمدا ومتى سٌتم تؽٌٌر قٌمة تلك المتؽٌرات المرفوعة وبذلك ٌمكنك التؤكد من أن برنامجك ٌتم تنفٌذه بصورة صحٌحة

 اسحخذاو جعببير نًذا ببنشكم الأيثم

 لتمكنك من Ifٌمكنك تمرٌر تعبٌر واحد كجسم لتعبٌر لمدا وقد تم تقدٌم كلمة محجوزة ثلاثٌة جدٌدة هً 2008 فً فٌجول باٌزٌك 

 كتابة تعابٌر شرطٌة ذات نوع كامل

Dim x = IF(condition, 10, 20) 

وهذا ٌعنً أنه فً المثال السابق ٌتتبع المترجم كلا فرعً .  فٌما عدا أنها آمنة ضد النوعIIF مشابهة لاستدعاء الوظٌفة Ifوالكلمة المحجوزة 

 سٌعٌد IIf ولكن استخدام integer هو x وبهذا فهو ٌطبق قواعد الاستدلال على النوع وٌقرر أن نوع Integer وٌعٌد Ifالكلمة المحجوز 

  فً تعبٌر لمداIfكما ٌمكنك استخدام . Objectالنوع 

Dim x = Function(c As Customer) _ 
        If(c.Age >= 18, c.Address, c.Parent.Address) 

 التً تمثل العنوان الحالً للزبون حٌث أن تعبٌر Address ٌتضمن تعرٌفها الخاصٌة Customerففً المثال السابق افترض أنه لدٌك فئة 

 فهو ٌعٌد 18 لتطبٌق الشرط على محدد الدخل فإن كان عمر الزبون مساوٌا أو أكثر من Ternary Expressionلمدا ٌستخدم التعبٌر الثلاثً 

 ثم Addressعنوانه وإلا فهو ٌعٌد عنوان والده وهنا ٌتم استخدام الاستدلال على النوع أٌضا وٌقوم المترجم بتحدٌد نوع تعبٌر لمدا لٌكون 

 .Address كدخل وٌعٌد Customer بالطرٌقة التً تمت مناقشتها سابقا حٌث ٌؤخذ النوع المفوض xٌقوم بإنشاء النوع المفوض 

Object Initializers 

 

 الأنواع من كلا من متؽٌرات لتعرٌؾ وتستخدم واحد تعبٌر ضمن معقد ؼرض خصائص تحدٌد من Object Initializers  تمكنك

 الشكل على معرفة Employee بسٌطة فئة لدٌنا أنه فرضنا فلو والمجهولة المعروفة

Public Class Employee 

    Private _name As String 

    Private _Salalry As Short 

    Private _Address As String 

 

    Public Property Name() As String 

        Get 

            Return _name 

        End Get 

        Set(ByVal value As String) 

            _name = value 

        End Set 

    End Property 

 

    Public Property Salary() As Short 

        Get 

            Return _Salalry 

        End Get 

        Set(ByVal value As Short) 

            If value > 0 Then 

                _Salalry = value 

            End If 

        End Set 

    End Property 

 

    Public Property Address() As String 

        Get 

            Return _Address 

        End Get 

        Set(ByVal value As String) 

            _Address = value 

        End Set 

    End Property 

End Class 

 كافة قٌم لضبط هنا مضطرٌن لسنا أننا مع التالً الكود فً كما الخصائص واسندا الفئة تلك إلى ٌشٌر متؽٌر تعرٌؾ باستخدام ٌمكننا

 فقط لضبطها نحتاج التً الخصائص قٌم بضبط فنقوم تحتوٌها الفئة التً الخصائص

Dim Empl3 = New Employee With {.Name = "Mazen", .Salary = 8500} 

Dim Empl1 As New Employee With {.Name = "Reem", .Salary = 10000} 

 local type inference على اعتماد وذلك ٌلً كما التصرٌح كتابة فٌمكن هنا AS قسم اختصار ٌمكننا كما

Dim Empl5 = New Employee With {.Name = "Ahmad"} 

 ٌلً كما الفئة نفس وباستخدام السابق فً كنا بٌنما

        Dim Empl2 As New Employee 

        With Empl2 

            .Name = "Ahamd" 

            .Salary = 11500 

        End With 

 لا ٌتم أخرى خصائص لضبط الطرٌقة نفس استخدام أٌضا مثلا فٌمكننا Person الفئة مثل الفئة لمشٌد قٌم لتمرٌر تحتاج فئة لدٌنا كانت وإن

 الفئة لمشٌد تمرٌرها

Dim Per1 As New Person("Ghassan") With {.Address = "Damas"} 

 المجهولة الأنواع لتعرٌؾ أٌضا الطرٌقة هذه تستخدم كما

Dim Visitor = New With {.Name = "Mussa", .Account = 232536} 

 اسم وجود لاحظ المعروفة الأنواع ففً المجهولة للأنواع الشكل فً مماثلة المعروفة الأنواع تعرٌؾ فصٌؽة التعرٌؾ طرٌقة من نلاحظ وكما

 قابلة فئة اسم لها لٌس المجهولة الأنواع أن بسبب new الكلمة بعد للفئة اسم لاٌوجد مجهولا نوعا نعرؾ عندما بٌنما newالكلمة  بعد الفئة

 متؽٌرا ٌنشئ والتصرٌح فعلا موجودة قٌمها ضبط نرٌد التً الخصائص تكون أن ٌجب التصرٌح عند معروفة فئة استخدام فعند للاستخدام

 فً إلٌها المشار الخصائص تحتوي المتؽٌر لذلك جدٌدة فئة بإنشاء المترجم المجهول ٌقوم النوع تعرٌؾ أجل ومن الفئة تلك إلى ٌشٌر

 الكود ضمن المجهولة الفئات اسم على الاعتماد لذلك لا ٌمكن لأخرى ترجمة عملٌة من لاسم ٌختلؾ وقد الترجمة عند اسمها وٌحدد التصرٌح

 التعرٌؾ أو

 بالتعرٌؾ الخاصة الملاحظات بعض وإلٌك

 فارؼة تكون أن لا ٌمكن With بعد التعرٌؾ قائمة- 

 التعرٌؾ نفس فً مرة من أكثر لخاصٌة قٌمة تعرٌؾ تكرار لا ٌمكن- 

 أخرى خاصٌة من خاصٌة قٌمة ضبط ٌمكن- 

 الطرٌقة بنفس التصرٌح تعشٌش ٌمكن فئة الخصائص إحدى كانت حال فً- 

Dim cust12 = New Customer With {.Name = "Toni Poe", _ 

                                .Address = New AddressClass _ 

                                With {.City = "Louisville", _ 

                                      .State = "Kentucky"}} 

Console.WriteLine(cust12.Address.State) 

 

 With كلمة بعد القائمة فً الطرق استدعاء أو الثوابت أو ReadOnly فقط للقراءة أو Shared مشتركة عناصر استخدام لا ٌمكن- 

 صحٌحة ؼٌر مثلا التالٌة فالتعرٌفات مثلا كمصفوفة المشروطة أو فهرسا تمتلك التً الخصائص استخدام لا ٌمكن- 

'' Not valid. 

' Dim c1 = New Customer With {.OrderNumbers(0) = 148662} 

' Dim c2 = New Customer with {.Address.City = "Springfield"} 

  نحهك انًشبريعLinq ثى إضبفة دعى 2008 نحعًم عهى 2005جرقية يشبريع 

 

  اضؽط – فٌظهر لك معالج الترقٌة تلقائٌا 2008افتح مشروعك ضمن بٌئة تطوٌر next 

  ٌظهر لك المعالج خٌار عمل نسخة احتٌاطٌة للملفات القدٌمة فنختار الخٌارYes, create a backup before converting ثم 

 حٌث ٌمكنك تؽٌٌره بالضؽط Location for backupٌقترح مكانا لوضع النسخة الاحتٌاطٌة فٌه فً مربع النصوص تحت الكلمة 

  للبدء بعملٌة الترقٌةFinishاضؽط هنا -  فٌظهر لك معلومات عن الترقٌة nextاضؽط -   أو تركه كما هو Browseعلى الزر 

  بعد الانتهاء تظهر لك نافذة تخبرك بانتهاء عملٌة الترقٌة وفٌها خٌارShow the conversion log when the wizard is 

closed وبتفعٌل هذا الخٌار ٌظهر لك تقرٌر عن عملٌة التحوٌل بعد إؼلاق المعالج هنا اضؽط Close 

  2005كما تجدر ملاحظة أن مشروعنا بعد التحوٌل حتى هذه النقطة مازال متوافقا مع بٌئة تطوٌر الـ 

  منSolution Explorer انقر بزر الفؤرة الٌساري نقرا مزدوجا على My Projectلفتح خصائصه  

  افتح صفحةCompile  وقم بضبط  الخٌار Option Infer إلى On الذي ٌخبر المعالج أن ٌستدل على نوع المتؽٌرات المحلٌة من 

ولمزٌد من المعلومات حول هذا الخٌار ٌمكنك قراءة - التعبٌر الذي ٌضبط قٌمة ذلك المتؽٌر إن لم نزوده بنوع ذلك المتؽٌر 

 Local Type Inference النوع على المحلً موضوعً فً المنتدى بعنوان الاستدلال

  انتقل لأسفل صفحةCompile ثم اضؽط الزر Advanced Compile Options الذي ٌظهر لنا صندوق حوار بخٌارات الترجمة 

 حتى Framework 2.0 والذي مازال مضبوطا على الـ Target frameworkالمتقدمة  حٌث نرى فً أسفل هذه النافذة الخٌار 

الآن حٌث ٌمكننا هذا الخٌار من كتابة كود ٌعمل على أي نسخة من نسخ الفرٌموورك الموجودة ضمن بٌئة تطوٌر واحدة فبدلا من 

 أصبح الآن بإمكانك عمل ذلك Framework على نفس الجهاز بهدؾ العمل مع أكثر من Visual Studioتنصٌب عدة نسخ من 

 فٌظهر لنا تحذٌر Ok ثم اضؽط على Framework 3.5 سنختار Linq ومن أجل تمكٌن 2008من داخل بٌئة تطوٌر واحدة هً 

 Yesبؤنه سٌتم إؼلاق وفتح المشروع تلقائٌا وأنه سٌقوم بحفظ أٌة تؽٌٌرات ؼٌر محفوظة تلقائٌا وهنا اضؽط  

  انقر بزر الفؤرة الٌساري نقرا مزدوجا علىMy Project لفتح خصائصه وانتقل للصفحة References  فنلاحظ أنه قد تم إضافة 

 Target Framework للمشروع وذلك لأننا قمنا بترقٌة Framework 3.5 من الـ system.core.dllمرجعا تلقائٌا للمكتبة 

 الذي Linq علٌنا إضافة بعض المراجع واستٌراد بعض مجالات الأسماء اعتمادا على مزود Linqومن أجل تمكٌن استعلامات 

 نحتاج لاستخدامه 

  فمن أجل إضافة مرجع لـLinq to Object نختار من القائمة أسفل Import Namespaces المجال System.linq الذي ٌمكننا 

 على الأؼراض المختلفة حٌث أصبح بإمكاننا كتابة الاستعلام التالً للحصول على أسماء الملفات فً Linqمن كتابة استعلامات 

 المجلد الحالً 
Dim MyFiles = From Files In My.Computer.FileSystem.GetFiles(CurDir()) _ 

              Select Files 

 
 إلى MyFiles أن المترجم قد ضبط نوع المتؽٌر On المضبوطة إلى Option Infer ونلاحظ أنه بسبب خاصٌة 

IEnumerable(Of  String) وذلك عند تمرٌر مشٌرة الفؤرة فوق المتؽٌر MyFiles 

  ومن أجل تمكٌن استعلاماتLinq للاستعلام من الـ Datasets سنحتاج لبعض الإجراءات الإضافٌة وهنا افتح خصائص My 
Project وعد للصفحة References واضؽط الزر Add ومن صفحة .net أضؾ مرجعا لـ 

System.Data.DataSetExtensions ثم اضؽط OK 

  سنحتاج الآن لإعادة تولٌد الـDataset التً نرٌد استخدامها مع استعلامات Linq ولعمل ذلك ننقر بزر الفؤرة الٌمٌنً على الـ 

Dataset المناسبة ثم نختار Run Custom Tool من القائمة وهنا تقوم بٌئة التطوٌر بإعادة كتابة كود الـ Dataset لتصبح 

 وهذا الذي ٌستدعً الحاجة لإضافة مجال الأسماء TypedTableBase تسمى Linqالجداول ضمنها موروثة من فئة داعمة لـ 

System.Data.DataSetExtensions وأصبح بإمكاننا الآن كتابة استعلامات من الـ datasetمثل الاستعلام  
 

 Dim MyCats = From category In Me.CategoryProductDataSet.Categories _ 

              Where category.CategoryName Like "C*" _ 

              Select category 

 
 بناء على جملة الاستعلام المرتبطة به تلقائٌا إلى MyCats ونلاحظ هنا أٌضا أن المترجم قد ضبط نوع المتؽٌر 

 EnumerableRowCollection  

  ولكتابة استعلاماتLinq to XML عد إلى صفحة References فً خصائص MyProject واضؽط الزر Add وأضؾ مرجعا 

 حتى نتمكن من كتابة System.Xml.Linq وقم باختٌار Imported namespaces ثم انتقل إلى القائمة System.Xml.Linqلـ 

 XMLاستعلامات من 

  كما ٌوجد مزود آخر ربما نرٌد استخدامه وهو مزودLinq to Sql حٌث ٌمكن إضافته بسهولة فقط قم باختٌار Add New Item 

 وهذا سٌقوم تلقائٌا بإضافة المراجع والاستٌرادات المناسبة لمشروعنا حٌث Linq to Sql Classes وقم بإضافة Projectمن قائمة 

 System.Data.Linq أنه قد تم إضافة مرجعا لـ MyProject فً خصائص Referencesسنلاحظ من صفحة 
 

 أو Dataset المختلفة فً مشروعنا بالإضافة إلى الاستعلام من Objects للاستعلام على الأؼراض Linqأصبح الآن بإمكاننا استخدام 

XML أو حتى Sql Databse 

Linq To Object اسحعلايبت وأسبسيبت Linq 

 

 أو الواجهة IEnumerable ٌمكننا الاستعلام من أؼراض دوت نٌت المختلفة طالما هً تدعم الواجهة Linq to Objectباستخدام مزود 

IEnumerable(T)ًفمثلا ٌمكننا كتابة استعلام للحصول على قائمة بالملفات الموجودة فً المجلد الحال  

Dim Files = From Fi In My.Computer.FileSystem.GetFiles(CurDir()) _ 

            Order By Fi 

 فً بداٌة الاستعلام لم نصرح عن نوع  المتؽٌر وذلك لأن المترجم هنا ٌستدل على Filesنلاحظ بداٌة أننا عندما قمنا بالتصرٌح عن المتؽٌر 

نوعه من التعبٌر الذي ٌضبط قٌمته وهنا أفترض أنك قد درست موضوعً بخصوص الاستدلال المحلً عن النوع وفً حالة الاستعلام 

 ستجد أن بٌئة التطوٌر قامت بضبط نوعه إلى Filesالسابق إن قمت بتمرٌر مإشر الفؤرة فوق المتؽٌر 

System.Linq.IOrderedEnumeable(Of String) باستخدام الاستدلال المحلً على النوع ونبدأ بكتابة الاستعلام بالقسم From حٌث 

 وفً استعلامنا هنا In ٌحصل على قٌمته من الكائن الذي ٌلً الكلمة Fiنحدد أنه لدٌنا متؽٌر 

My.Computer.FileSystem.GetFiles(CurDir()) حٌث تجدر الملاحظة إلى أن استعلامات Linq جمٌعها تكون سطرا واحدا فً لؽة 

ونرٌد هنا أن نخرج _ فٌجول باٌزٌك لذا من أجل التنسٌق وقابلٌة القراءة والتعدٌل نقسم العبارة على عدة أسطر باستخدام محرؾ المتابعة 

 لٌقوم بترتٌب أسماء الملفات المعادة من الاستعلام وكً نظهر قائمة الملفات هذه فً OrderByقائمة مرتبة بؤسماء الملفات لذا نستخدم قسم 

ListBox نستخدم حلقة For … Each للدوران عبر عناصر المجموعة المعادة من الاستعلام بما أنها تحقق الواجهة IEnumerable 

 وإضافتها عنصرا عنصرا لمربع القائمة كما فً المثال

For Each f In Files 

    Me.ListBox1.Items.Add(f) 

Next 

كما ٌمكننا استخدام الاستعلام ضمن استعلام آخر فللحصول على معلومات الملفات المعادة من الاستعلام السابق ٌمكننا كتابة الاستعلام التالً 

 GetFileInfo بحسب القٌمة المعادة من الدالة FileInfo أن النتٌجة المعادة هً مجموعة من Selectحٌث ٌحدد القسم 

Dim FInfo = From File In Files _ 

            Select My.Computer.FileSystem.GetFileInfo(File) 

 ToList إلى نتٌجة عائد الدالة DataSource مباشرة وذلك بضبط قٌمة الخاصٌة DataGridViewوٌمكننا إظهار نتٌجة هذا الاستعلام فً 

  كما فً الكودFileInfoالخاصة بمتؽٌر الاستعلام 

Me.DataGridView1.DataSource = FInfo.ToList 

 التً نستخدمها فً Select لتخصٌص المعلومات المعادة من الاستعلام وبشكل مشابه لما كنا نفعله فً عبارة Selectكما ٌمكننا استخدام قسم 

 كما فً الاستعلام السابق ٌمكننا كتابة استعلامنا لإظهار اسم الملؾ ووقت الإنشاء FileInfo فبدلا من إعادة كافة خصائص SQLاستعلامات 

 فقط كما فً المثال

Dim MyInfo = From Fi In FInfo _ 

             Select Fi.Name, Fi.CreationTime 

 

كما ٌمكننا إعادة نتٌجة هذا الاستعلام ضمن ؼرض من إنشائنا بدلا من النوع الذي ٌتم تحدٌده تلقائٌا كنتٌجة للاستعلام فإن كان لدٌنا فئة بسٌطة 

  ٌمكننا عندها إعادة كتابة استعلامنا بالشكلString من النوع Name و Date من النوع CreationTime تمتلك خاصٌتٌن MyFilesباسم 

Dim MyFiles = From Fi In FInfo _ 

              Select New MyFiles() With {.Name = Fi.Name, .CreationTime = Fi.CreationTime} 

 

 بخصوص Object Initializers وٌمكنك مراجعة موضوعً IEnumerable(Of MyFiles)حٌث سٌعٌد الاستعلام النتٌجة كمجموعة  

  ضمن الاستعلامMyFilesصٌؽة تعرٌؾ الفئة 

 الذي ٌحدد شرط الانتقاء where فقط مثلا نستخدم قسم exeوإن أردنا تخصٌص ناتج الاستعلام للحصول على الملفات التً تحمل الامتداد 

  عندها ٌمكننا كتابة الاستعلام بالشكل التالexeً. لاختٌار اسم الملؾ الذي ٌنتهً بـ EndsWithفً جملة الاستعلام مستخدمٌن الطرٌقة 

Dim ExeFiles = From Fi In My.Computer.FileSystem.GetFiles(CurDir()) _ 

               Where Fi.EndsWith(".exe") _ 

               Select My.Computer.FileSystem.GetFileInfo(Fi) 

 

  ٌمكننا استخدام ناتج الاستعلام السابق فً استعلام جدٌدexeوإن أردنا الحصول على مجموع حجوم الملفات من النوع 

Dim ExeSize = Aggregate Fs In ExeFiles _ 

              Into Sum(Fs.Length) 

 

Me.TextBox1.Text = ExeSize 

 فً بداٌة الاستعلام عندما نرٌد استخدام الدالات التجمٌعٌة للحصول على نتائج تجمٌعٌة من From بدلا من Aggregateفهنا استخدمنا 

  للحصول على مجموع الحجومSum استخدمنا الدالة Intoالاستعلام ففً قسم 

 وقمنا بإنشاء قائمة تحتوي  Salary و City و Age و Birthdate و Name تمتلك الخصائص Personnelإذا افترضنا أنه لدٌنا فئة باسم 

 على مجموعة عناصر تمتلك نوع هذه الفئات 

Dim Pers As New List(Of Personnel) 

وبافتراض أن هذه القائمة تحتوي على العدٌد من العناصر ٌمكننا كتابة مجموعة من الاستعلامات للحصول على معلومات مختلفة حول 

 وأردنا فً الناتج فقط الاسم والعمر والراتب وترتٌب النتائج 10000عناصر هذه القائمة فإن أردنا قائمة بالأشخاص الذٌن راتبهم أكثر من 

 بحسب الراتب ٌمكننا كتابة الاستعلام كما ٌلً 

Dim Prs = From p In Pers _ 

          Where p.Salary > 10000 _ 

          Order By p.Salary _ 

          Select p.Name, p.Age, p.Salary 

 

 وإن أردنا فقط الأشخاص الذٌن ٌقٌمون فً مدٌنة دمشق فقط سٌصبح استعلامنا بالشكل

Dim Prs = From p In Pers _ 

          Where p.Salary > 10000 And p.City = "Damascus" _ 

          Order By p.Salary _ 

          Select p.Name, p.Age, p.Salary 

 

  ٌصبح استعلامنا كما ٌل10000ًوإذا أردنا الأشخاص فً مدٌنتً دمشق وحماة الذٌن رواتبهم أكثر من 

Dim Prs = From p In Pers _ 

          Where p.Salary > 10000 _ 

          And (p.City = "Damascus" Or p.City = "Hama") _ 

          Order By p.Salary _ 

          Select p.Name, p.Age, p.Salary, p.City 

 

 وإن أردنا الحصول على مجموع رواتب الأشخاص المقٌمٌن فً حلب ٌمكننا كتابة الاستعلام

Dim prs = Aggregate p In Pers _ 

          Where p.City = "Aleppo" _ 

          Into Sum(p.Salary) 

 وإن أردنا الحصول على معلومات الشخص الذي ٌحصل على اعلى راتب 

Dim pr = From p In Pers _ 

         Aggregate pa In Pers _ 

         Into a = Max(pa.Salary) _ 

         Where p.Salary = a _ 

         Select p 

 وإن أردنا معلومات من ٌحصل على أقل راتب فً مدٌنة حلب

Dim pr = From p In Pers _ 

         Aggregate pa In Pers _ 

         Where pa.City = "Aleppo" _ 

         Into a = Min(pa.Salary) _ 

         Where p.Salary = a And p.City = "Aleppo" _ 

         Select p 

 

 عناصر مجموعة على تحتوي قائمة بإنشاء وقمنا City و BranchName تمتلك الخصائص Branchesوبافتراض انه لدٌنا فئة  ثانٌة باسم 

 الفئات هذه نوع من

Dim Brnch As New List(Of Branches) 

 ٌمكننا كتابة الاستعلام التالً للحصول على اسم الشخص والفروع المتوفرة فً مدٌنته

Dim BrnPer = From pe In Pers _ 

             Join br In Brnch On pe.City Equals br.City _ 

             Select PersonName = pe.Name, BranchName = br.BranchName, pe.City 

 لتحدٌد الحقول المطلوب إخراجها Select لربط مجموعة الأشخاص مع مجموعة الفروع باستخدام المدٌنة ثم استخدمنا joinحٌث استخدمنا 

 فً نتٌجة الاستعلام

 الآن نرٌد إظهار قائمة بجمٌع الأشخاص مع الفروع المتوفرة لهم وبذلك سٌصبح استعلامنا بالشكل

Dim PreBr = From pe In Pers _ 

            Group Join br In Brnch On pe.City Equals br.City _ 

            Into AvBr = Group _ 

            Select pe, AvBr 

 

 لربط هذه القائمة مع قائمة الفروع باستخدام حقل المدٌنة Group Join ثم استخدمنا Pers من قائمة الأشخاص Peفً البداٌة اخترنا المتؽٌر 

  AvBr و Pe النتائج التً نرٌدها Select ثم حددنا فً قسم Into فً قسم AvBrثم وضعنا ناتج الربط من القائمة الثانٌة فً متؽٌر 

  متداخلتانFor … Each باستخدام حلقتً ListBoxوٌمكننا إظهار ناتج الاستعلام فً 

For Each a In PreBr 

    Me.ListBox1.Items.Add(a.pe.Name & ": " & a.pe.City) 

    For Each b In a.AvBr 

        Me.ListBox1.Items.Add(".... " & b.BranchName) 

    Next 

Next 

 

 ولعمل ذلك فً BindingSource و تحكمان DataGridView لإظهار النتائج سنحتاج إلى تحكمان DataGridViewأو إذا أردنا استخدام 

  حٌث تكون المفاتٌح هً الأشخاص والقٌم هً الفروع كما ٌلDictionaryًالبداٌة سنعرؾ متؽٌرا على مستوى النموذج من النوع 

Private Gper As Dictionary(Of Personnel, IEnumerable(Of Branches)) 

لإضافة المفاتٌح -  هل درست المواضٌع المتعلقة بتعابٌر لمدا – حٌث نستخدم تعابٌر لمدا Gperثم نقوم بوضع ناتج الاستعلام فً متؽٌرنا 

 والقٌم إلٌه كما فً الكود

Gper = PreBr.ToDictionary(Function(x) x.pe, Function(y) y.AvBr) 

  كما فً الكودDataGridView مع تحكمات BindingSourceثم سنقوم بربط تحكمات 

Me.DataGridView1.DataSource = Me.BindingSource1 

Me.DataGridView2.DataSource = Me.BindingSource2 

  Gper إلى قٌم مفاتٌح BindingSource1 لـ DataSourceثم نقوم بضبط خاصٌة 

Me.BindingSource1.DataSource = Gper.Keys 

 التً ستظهر قائمة الأشخاص ولإظهار قائمة الفروع المتوفرة لكل شخص فً DataGridView1وبذلك ٌتم إظهار قٌم المفاتٌح فً 

DataGridView1 ًف DataGridView2 نقوم بمعالجة الحدث CurrentChanged لـ BindingSource1ًباستخدام سطر الكود التال  

Me.BindingSource2.DataSource = Gper(CType(Me.BindingSource1.Current, Personnel)) 

 لـ DataSource وٌضبطها كـ BindingSource1 المقابلة للسجل الحالً فً Dictionaryالذي ٌحصل على القٌمة فً الـ 

BindingSource2 ًفٌتم عرضها ف DataGridView2 ًالتً ستظهر قائمة الفروع المتوفرة لذلك الشخص عند النقر علٌه ف 

DataGridView1 

Linq To DataSet 

 

 فً بٌئة التطوٌر أضؾ مصدر بٌانات جدٌد Data Sources مثبتة فً جهازك من نافذة Northwindبافتراض أن قاعدة البٌانات 

 إلى سطح النافذة Categories وباستخدام طرٌقة السحب والإفلات اسحب الجدول Products و Categoriesلمشروعك ٌتضمن الجدولٌن 

 Categories وسع العقدة بجانب الجدول DataSources مع بعض التحكمات للنموذج ثم من نافذة DataGridViewالفارغ لإضافة 

 ثانٌة أسفل الأولى خاصة DataGridView إلى سطح النافذة لٌتم إنشاء Categories الذي بداخل الجدول Productsواسحب الجدول 

 فإن واجهت Linq طبعا لن أقوم بشرح هذه العملٌة بتفصٌل أكثر بما أنها برمجة قواعد بٌانات ودورتنا تتحدث عن Productsبالجدول 

مشكلة ابحث فً القسم المناسب فً المنتدى أو حاول رإٌة فٌدٌوهات مٌكروسوفت التعلٌمٌة بخصوص هذه النقطة وقبل المتابعة ٌجب أن 

 . الثانٌة تعرض البٌانات المقابلة لما تم اختٌاره من الأولى فقطDataGridViewتتؤكد أن مشروعك ٌعمل جٌدا وأن الـ 

 

 وللحصول CategoriesBindingSource للتحكم CurrentChangedانتقل إلى محرر الكود للنموذج وقم بإنشاء إجراء معالجة للحدث 

 على السطر الحالً نستخدم الكود التالً

Dim row As NorthwindDataSet.CategoriesRow 

row = CType(CType(Me.CategoriesBindingSource.Current,  _ 

      DataRowView).Row, NorthwindDataSet.CategoriesRow) 

 

 وللحصول على إجمالً قٌمة البضائع فً تلك الفئة والتً مازال إنتاجها مستمرا ٌمكننا كتابة الاستعلام

Dim Total = Aggregate Product In NorthwindDataSet.Products _ 

            Where Product.CategoryID = row.CategoryID _ 

            AndAlso Product.Discontinued = False _ 

            Into Sum(Product.UnitPrice * Product.UnitsInStock) 

 

 فً بداٌة الاستعلام بما أننا نرٌد استخدام Aggregateوبافتراض أنك متابع معً منذ البدء أصبحت تعرؾ طرٌقة الاستعلام فهنا استخدمنا 

 الشرط بؤننا نرٌد أن Whereالدالات التجمٌعٌة للحصول على مجموع الكلفة الإجمالٌة للبضائع التً مازال إنتاجها مستمرا ثم حددنا فً قسم 

 وأن إنتاجها ما زال مستمرا Product.CategoryID = row.CategoryIDٌتم تجمٌع المنتجات الموافقة للفئة المحددة 

Product.Discontinued = False ثم استخدمنا قسم Into للحصول على الإجمالً المطلوب وذلك بتمرٌر جداء قٌمة المنتج والعدد 

ولإظهار نتٌجة الاستعلام على النموذج ضع صندوق نصوص على النموذج واستخدم الكود التالً لوضع . Sumالموجود للدالة التجمٌعٌة 

  لتنسٌق القٌمة المعادة من الاستعلام بتنسٌق عملةFormatالقٌمة فٌه الذي ٌستخدم الأمر 

Me.TextBox1.Text = Format(Total, "c") 

لتنفٌذ بعض التصفٌة على الفئات أضؾ تحكم صندوق نصوص وزر إلى شرٌط الأدوات الموجود فً أعلى النموذج وفً إجراء حدث النقر 

 على الزر أدخل الاستعلام التالً

Dim SelCat = From Cat In Me.NorthwindDataSet.Categories _ 

           Where Cat.CategoryName.ToLower Like  Me.ToolStripTextBox1.Text.ToLower & "*" _ 

           Select Cat 

 

Me.CategoriesBindingSource.DataSource = SelCat.AsDataView  

 لتصفٌة النتائج المعادة من الاستعلام تماما كما نفعل فً استعلامات قواعد البٌانات ثم نقوم بضبط Where فً قسم Likeحٌث قمنا باستخدام 

 الخاصة AsDataView لكً ٌظهر لنا نتائج الاستعلام حٌث أن الطرٌقة CategoriesBindingSource لـ DataSourceخاصٌة 

 Linq to DataSet ٌمثل استعلام Linq داعم لـ DataViewبالاستعلام تعٌد ؼرض 

 وزر أوامر جدٌدان علٌها ثم سنستخدم كودنا DataGridViewدعنا نجري بعض الاستعلامات الأخرى وسع نافذة مشروعك قلٌلا وأضؾ 

 للحصول على معلومات السجل الحالً الذي تم CategoriesBindingSourceالسابق الذي ٌحصل على متؽٌر ٌحمل السجل الحالً فً 

  لزر الأوامرClick وذلك فً بداٌة إجراء معالجة الحدث CategoriesDataGridViewاختٌاره فً  

Dim row As NorthwindDataSet.CategoriesRow 

row = CType(CType(Me.CategoriesBindingSource.Current,  _ 

      DataRowView).Row, NorthwindDataSet.CategoriesRow) 

 

الآن ٌمكننا كتابة الاستعلام التالً للحصول على قائمة بالمنتجات التً مازالت قٌد الإنتاج مع السعر الإجمالً للموجود منها حالٌا وإظهار 

  والتً من الفئة التً تم اختٌارها من الشبكة الخاصة بالفئات DataGridView1النتٌجة فً 

Dim Prods = From pr In NorthwindDataSet.Products _ 

            Where pr.CategoryID = row.CategoryID _ 

            And pr.Discontinued = False _ 

            Select pr.ProductName, pr.UnitPrice, _ 

            pr.UnitsInStock, Total = (pr.UnitPrice * pr.UnitsInStock) 

 

Me.DataGridView1.DataSource = Prods.ToList 

 

أضؾ ثلاث صنادٌق نصوص إلى النافذة حٌث سنقوم بإنشاء استعلام جدٌد لحساب متوسط سعر الوحدات الموجودة ومجموع الكمٌات والقٌمة 

 الإجمالٌة وذلك من أجل نفس الفئة التً أظهرنا نتائجها فً الاستعلام السابق 

Dim ProdSums = Aggregate pr In NorthwindDataSet.Products _ 

               Where pr.CategoryID = row.CategoryID _ 

               And pr.Discontinued = False _ 

               Into UntiSum = Sum(pr.UnitsInStock), PriceAvg = Average(pr.UnitPrice), _ 

               TotalValue = Sum(pr.UnitPrice * pr.UnitsInStock) 

 

Me.TextBox2.Text = ProdSums.PriceAvg.ToString("#,###.00") 

Me.TextBox3.Text = ProdSums.UntiSum 

Me.TextBox4.Text = ProdSums.TotalValue 

 

 مع بعض الاختلاؾ فً الترتٌب وأن صٌؽة SQL فً select قرٌبة جدا من طرٌقة كتابة استعلامات Linqلاحظ أن طرٌقة كتابة استعلامات 

 الذي نتعامل معه حٌث ٌمكننا استخدام الأشكال المختلفة للاستعلام التً وردت فً الدرس Linqهذه الاستعلامات متشابهة مهما اختلؾ مزود 

 Linq to وفً الدروس المستقبلٌة عندما نتحدث عن Linq to Dataset هنا أٌضا عندما ٌتعلق الأمر بـ Linq to Objectالسابق المتعلق بـ 

xml و Linq to Sql 

 Lambda Expressions يع اسحخذاو Linq To DataSetيثبل عًهي عهى 
 

 الهدؾ من المثال

  واستعادتها منه والتعامل معها بدون الحاجة لوجود قاعدة بٌانات xml فً ملؾ DataSetحفظ بٌانات  .1

 ومن أكثر من جدول وإدخال بعض الحسابات فً جملة الاستعلام واستخدام عبارة Linq باستخدام DataSetالاستعلام من الـ  .2

Joinلمنع التكرار الخاطئ للبٌانات  

  للقٌام بالحسابات من أجلنا والاستفادة من مٌزة رفع المتؽٌراتLambda Expressionsاستخدام  .3

  مباشرةDataGridView فً Linqإظهار نتٌجة استعلام  .4

 ٌمكن ربطها مع أي قاعدة بٌانات سؤقوم بالعمل على مجموعة DataSet من أجل ترك الموضوع عام وبما أن مجموعة البٌانات 

 وسٌكون DataSetبٌانات ؼٌر مربوطة مع قاعدة بٌانات حٌث ٌمكنك تطبٌق الأفكار الواردة هنا على أي قاعدة بٌانات ٌمكن ربطها مع أي 

 مثالً الذي سنسٌر علٌه هنا معتمد على خدمة مصرفٌة تدعى بالودائع لأجل

 اختر Project ثم من قائمة Windows Forms Application وأنشئ مشروعا جدٌدا من نوع 2008افتح أي إصدار من فٌجول ستودٌو 

 ثم فً محرر التصمٌم الرسومً لمجموعة البٌانات انقر MyDataSet للمشروع وقم بتسمٌتها DataSet وأضؾ Add New Itemالأمر 

 ثم أضؾ Customers فً قائمة السٌاق ثم قم بإعادة تسمٌته لٌصبح اسمه Add من قائمة DataTableبزر الفؤرة الٌمٌنً واختر الأمر 

  الحقول التالٌة مع الخصائص الموضحة بجانب كل منهاCustomersللجدول 

ى إل كلتاهما AutoIncrementStep  و AutoIncrementSeed والخاصٌتان True بالقٌمة AutoIncrement الخاصٌة IDالحقل 

 من قائمة السٌاق Set Primary Key واختر الأمر ID ثم انقر بزر الفؤرة الٌمٌنً على الحقل System.Int32 و نوع البٌانات 1القٌمة 

 لتحدٌد الحقل كمفتاح أساسً

  25 بالقٌمة MaxLength و System.String نوع البٌانات CustomerNameالحقل 

 MaxLength 25 و System.String البٌانات نوع CurrentAccountNumber الحقل

 هو رقم الحساب CurrentAccountNumberالزبون والحقل  اسم هو  CustomerName و الزبون معرؾ هو ID الحقل أن حٌث

 الجاري لدى المصرؾ

 منها كل بجانب الموضحة الخصائص مع وأضؾ له الحقول التالٌة Wadaeaaأضؾ جدول آخر لمجموعة البٌانات باسم 

 String بنوع بٌانات WadeaaNumberالحقل 

 Int32 بنوع بٌانات CustomerIDالحقل 

 7.5 مساوٌة لـ NullValue و Decimal بنوع بٌانات InterestRateالحقل 

 3 مساوٌة لـ BullValue و Int16 بنوع بٌانات WadeeaPeriodالحقل 

 DateTime بنوع بٌانات StartDateالحقل 

 10000 بقٌمة NullValue و Int32 بنوع بٌانات WadeaaAmountالحقل 

 نسبة الفائدة و InterestRate هو حقل مرتبط بجدول الزبائن و CustomerID هو رقم الودٌعة و WadeaaNumberحٌث 

WadeeaPeriod فترة الودٌعة بالأشهر و StartDate تارٌخ فتح الودٌعة و WadeaaAmountقٌمة الودٌعة  

 ثم اضبط Relation اختر Add ومن القائمة الفرعٌة Customersانقر بزر الفؤرة الٌمنً على الجدول : سنضٌؾ الآن علاقة بٌن الجدولٌن

Parent Table إلى Customers و Child Table إلى Wadaeaa ثم اختر Key Columns لٌضم الحقل ID فقط و Foreign Key 

Columns لٌضم الحقل CustomerID فقط ثم اختر الخٌار Both Relations And Foreign Key Constraint ثم اضؽط Ok من أجل 

 File من القائمة Save Allحفظ العلاقة ثم اختر الأمر 

 مع شرٌط أدوات وبعض DataGridView × 2 كبٌرة كفاٌة لتتسع لـ  Form1 واجعل مساحة Form1انتقل إلى محرر النماذج الخاص بـ 

 فٌتم Form1 ثم ألقه على سطح Customers واسحب الجدول Data Sourcesالتحكمات الأخرى التً سنضٌفها لاحقا ثم انتقل لنافذة 

  و شرٌط أدوات للنافذةDataGridViewإضافة 

 لتظهر لك قائمة بالحقول الخاصة به كما نلاحظ وجود نسخة من  Customersبجانب الجدول +  انقر إشارة  Data Sourcesمن نافذة 

 وذلك بسبب العلاقة التً قمنا بإنشائها بٌن الجدولٌن الآن قم بسحب الجدول Customers كجدول فرعً ضمن Wadaeaaالجدول 

Wadaeaa الموجود كجدول فرعً لـ Customers ولٌس الجدول الخارجً إلى سطح Form1 لٌتم إنشاء DataGridView أخرى على 

  فً الأعلى و الأخرى فً الأسفل Customers الخاصة بـ DataGridViewالنافذة ثم قم بتنسٌق النافذة بشكل جٌد وتؤكد من أن الـ 

 واضبط Edit Columns وانقر على السهم الصؽٌر الذي ٌظهر فً زاوٌتها الٌمٌنٌة العلٌا واختر الأمر CustomersDataGridViewاختر 

 WadaeaaDataGridView فً CustomerID ثم كرر العملٌة بالنسبة للحقل False إلى ID للحقل Visibleالخاصٌة 

 لتفعٌله ثم انقر علٌه Enabled واختر الأمر – ٌمتلك أٌقونة قرص –فً شرٌط الأدوات فً الأعلى انقر بزر الفؤرة الٌمٌنً على زر الحفظ 

 لتخزٌن WriteXmlنقرا مزدوجا لننتقل إلى محرر الكود ثم أدخل الكود التالً فً حدث النقر على زر الحفظ حٌث نستخدم الوظٌفة 

 xmlمحتوٌات مجموعة البٌانات فً ملؾ 

Try 

    MyDataSet.WriteXml("d:\TestData.xml") 

Catch ex As Exception 

    MsgBox(ex.Message) 

End Try 

 

 لاحظ ظهور رسالة خطؤ عند xml للنموذج وأدخل فٌه الكود التالً الذي سٌقوم بتحمٌل البٌانات من ملؾ Load أنشئ إجراء لمعالجة الحدث 

 من أجل Try … Catchتشؽٌل البرنامج لأول مرة وقبل حفظ البٌانات حٌث أن ملؾ البٌانات لم ٌتم إنشاإه بعد وهذا السبب فً استخدام بلوك 

  إلى مجموعة البٌاناتxml لتحمٌل البٌانات من ملؾ ReadXmlاصطٌاد الخطؤ وتجنب إفشال عملٌة بدء البرنامج وقد استخدمنا الوظٌفة 

Try 

    MyDataSet.Clear() 

    MyDataSet.ReadXml("d:\TestData.xml") 

Catch ex As Exception 

    MsgBox(ex.Message) 

End Try 

 

 وتؤكد من أنك قد قمت بتعبئة جمٌع الحقول فً كلتا شبكتً البٌانات DataGridViewشؽل البرنامج وأدخل فٌه بعض البٌانات فً كلا الـ 

 الحقل لهذا كقٌمة 12 أو 6 أو 3 أو 1 بالقٌم الالتزام حاول لذا الودٌعة فترة أشهر عدد عن عبارة هو WadeeaPeriod الحقل أن إلى وانتبه

وقم بالحفظ وأؼلق البرنامج ثم أعد فتحه من جدٌد للتؤكد من أن عملٌة الحفظ قد تمت بشكل صحٌح لاحقا  الاحتساب إجرائٌة تجربة أجل من

 لاحظ عدم ظهور رسالة الخطؤ التً ظهرت عند فتح البرنامج لأول مرة بعد أن أدخلنا بٌانات وقمنا بحفظها عند تشؽٌل البرنامج للمرة الثانٌة 

 نرٌد الآن إظهار قٌم الاحتسابات الخاصة بكل ودٌعة عند المرور علٌها وكذلك تارٌخ استحقاق هذه الودٌعة

 لقٌمة txtInterest لتارٌخ الاستحقاق و txtEndDateأضؾ أربعة حقول نصٌة للنافذة ورتبها أسفل شبكتً البٌانات وأعطها الأسماء التالٌة 

 لضرٌبة الإدارة المحلٌة ثم انتقل لمحرر الكود وأضؾ الاستٌراد التالً فً بداٌة ملؾ الكود txtIdara لضرٌبة الرٌع و txtRayaaالفائدة و 

 Math من أجل تمكٌننا من استخدام الوظائؾ الموجودة فً مجال الأسماء Form1الخاص بـ 

Imports ma = System.Math 

 لأنها الأساس فً  Lambda Expressionsثم أضؾ الإجراء التالً كإجرائٌة للاحتساب وهنا أرجو أن تكون قد تابعت دروسً المتعلقة بـ 

 إجرائٌة الاحتساب

Private Sub DisplayWaeaaCalcs(ByVal Amount As Integer, ByVal StartDate As Date, _ 

       ByVal EndDate As Date, ByVal Interest As Decimal) 

 

        Dim DaysNum = DateDiff(DateInterval.Day, StartDate, EndDate) 

        Dim Rayaa As Decimal 

 

        Dim Rayya_Calc = Function(Intrst As Decimal) ma.Ceiling(Intrst * 7.5 / 100) 

        Dim Idara_Calc = Function() ma.Ceiling(Rayaa * 10 / 100) 

        Dim Intrst_Calc = Function() _ 

                             ma.Ceiling(Amount * DaysNum * Interest / 36500) 

 

        Dim Intr = Intrst_Calc() 

        Rayaa = Rayya_Calc(Intr) 

        Dim Ida = Idara_Calc() 

 

        Me.txtEndDate.Text = EndDate.ToString("dd/MM/yyyy") 

        Me.txtInterest.Text = Intr.ToString("#,###.00") 

        Me.txtRayaa.Text = Rayaa.ToString("#,###.00") 

        Me.txtIdara.Text = Ida.ToString("#,###.00") 

    End Sub 

 للحصول على عدد أٌام الفترة التً سنقوم بالاحتساب عنها وكنا قد مررنا قٌم المبلػ و تارٌخ DateDiffحٌث استخدمنا فً البداٌة الدالة 

 بناء Rayya_Calcالبداٌة وتارٌخ النهاٌة ونسبة الفائدة كمحددات لإجرائٌة الاحتساب ثم عرفنا تعبٌر لمدا ٌقوم باحتساب قٌمة ضرٌبة الرٌع 

على مبلػ الفائدة الممررة له وفً تعبٌر لمدا والضرٌبة الأخرى لم نمرر لها قٌمة ولكنها استخدمت متؽٌر محلً من أجل الاحتساب وهنا 

 Intrst_Calcأنصحك بمراجعة قسم رفع المتؽٌرات فً مواضٌعً المتعلقة بتعابٌر لمدا إن لم تكن قد قرأته حتى الآن وتعبٌر لمدا الأخٌر 

ٌستخدم أٌضا خاصٌة رفع المتؽٌرات ولكنه هنا ٌستخدم المحددات الممررة للإجراء كمتؽٌرات مرفوعة ثم نقوم باستخدام هذه التعابٌر 

 للاحتساب ثم نظهر القٌم فً صنادٌق النصوص المناسبة

 لكلا شبكتً البٌانات CellEnterمن أجل أن نقوم باحتساب القٌم الموافقة لكل ودٌعة عند المرور علٌها سنقوم بعمل إجراء معالجة للحدث 

 فً بداٌة تعرٌؾ جسم Handles لكلا الشبكتٌن لاحظ ما بعد عبارة CellEnter أدخل الكود التالً كإجراء لمعالجة الحدث –بإجراء واحد 

 الإجراء وأٌضا أننً لم أقم بتمرٌر أٌة محددات لإجراء معالجة الحدث حٌث ٌمكننً فعل ذلك بما أننً متؤكد من أننً لن أحتاج لاستخدامها

Private Sub WadaeaaDataGridView_CellEnter() _ 

                        Handles WadaeaaDataGridView.CellEnter, CustomersDataGridView.CellEnter 

 

        Try 

            Dim EdDat = From a In MyDataSet.Wadaeaa _ 

                Where a.CustomerID = Me.CustomersDataGridView.CurrentRow.Cells(0).Value _ 

                And a.WadeaaNumber = Me.WadaeaaDataGridView.CurrentRow.Cells(0).Value _ 

                Select a.WadeaaNumber, a.WadeaaAmount, a.StartDate, _ 

                EndDate = DateAdd(DateInterval.Month, a.WadeeaPeriod, a.StartDate), _ 

                a.InterestRate 

 

            If EdDat.Count > 0 Then 

                DisplayWaeaaCalcs(EdDat.First.WadeaaAmount, EdDat.First.StartDate, _ 

                                  EdDat.First.EndDate, EdDat.First.InterestRate) 

            End If 

        Catch ex As Exception 

            Me.txtEndDate.Text = String.Empty 

            Me.txtInterest.Text = String.Empty 

            Me.txtRayaa.Text = String.Empty 

            Me.txtIdara.Text = String.Empty 

        End Try 

    End Sub 

 هو كٌان من الجدول a للحصول على القٌم الخاصة بالودٌعة التً نقؾ علٌها حٌث أن المتؽٌر Linqفً البداٌة قمنا بإنشاء استعلام 

Wadaeaa ثم فً قسم Where  ضبطنا الشرط بحٌث ٌجلب الاستعلام فقط الودائع الخاصة بزبون معٌن عن طرٌق التؤكد من أن قٌمة 

 الخاصة بالزبون من خلال قراءة قٌمة الخلٌة المناسبة فً السطر الحالً وتتمة للشرط وبنفس ID مساوٌة لقٌمة  CustomerIDالحقل 

 لنحدد فٌه قائمة الحقول التً نرٌد الحصول علٌها لاحظ Selectالطرٌقة قمنا بضبط الشرط كً ٌجلب الودٌعة ذات الرقم المراد ثم ٌؤتً قسم 

 التً تضٌؾ فترة DateAdd الذي ٌتم حساب قٌمته من الحقول المعادة من الاستعلام باستخدام الوظٌفة  EndDateوجود الحقل المحسوب 

زمنٌة معٌنة حسب المحددات الممررة لها إلى تارٌخ ممرر لها وتعٌد قٌمة التارٌخ الجدٌد وتعاد قٌمته مع قائمة الحقول التً ٌعٌدها الاستعلام 

 للقٌام DisplayWaeaaCalcs ثم نستدعً الوظٌفة  Countوبعد الاستعلام نتؤكد من أنه قد جلب نتائج فعلا بالتحقق من قٌمة الخاصٌة 

 بالحسابات وإظهار النتائج

شؽل البرنامج ولاحظ ظهور قٌم الاحتسابات فً مربعات النصوص عند التنقل فً كلا شبكتً البٌانات إذا كانت لدٌك بٌانات قمت بحفظها كما 

 طلبت منك سابقا

 له واضبط DataGridView ثم أضؾ Form2من أجل إظهار الودائع التً تبدأ بتارٌخ معٌن وإظهارها أضؾ نموذج آخر للمشروع باسم 

 لجعل شبكة البٌانات تملأ كامل مساحة النموذج ثم نسق النموذج بحٌث ٌكون كبٌرا كفاٌة لعرض البٌانات الناتجة Fill للقٌمة Dockالخاصٌة 

  واجعل إجراء معالجة حدث النقر علٌه ٌماثل الكود التالFrom1ًعن الاستعلام ثم أضؾ زرا للنموذج 

    Private Sub Button1_Click() Handles Button1.Click 

        Dim d As Date = CDate(InputBox("Enter Date")) 

        Dim Dawa = From a In MyDataSet.Wadaeaa _ 

                  Join B In MyDataSet.Customers On _ 

                  a.CustomerID Equals B.ID _ 

                  Where a.StartDate = d _ 

                  Select B.CustomerName, B.CurrentAccountNumber, _ 

                  a.WadeaaNumber, a.WadeaaAmount, a.WadeeaPeriod, a.StartDate, _ 

                  EndDate = DateAdd(DateInterval.Month, a.WadeeaPeriod, a.StartDate) 

 

        Dim c As New Form2 

        c.DataGridView1.DataSource = Dawa.ToList 

        c.DataGridView1.Update() 

        c.ShowDialog() 

    End Sub 

 

لاحظ أننً استخدمت مٌزة جدٌدة فً فٌجول ستودٌو تمكننً من حذؾ محددات إجراء معالجة حدث ما إن كنت على ٌقٌن أننً لن أحتاج 

 للربط بٌن الجداول عند عملٌة الاستعلام كً نتجنب مشكلة ظهور بٌانات مكررة Joinلاستخدامها وفً جملة الاستعلام تلاحظ أننً استخدمت 

من أحد الجداول من أجل جمٌع سطور الجدول الآخر حٌث استخدمنا نفس أسلوب العلاقة التً قمنا بإنشائها فً البداٌة بٌن الجدولٌن من حٌث 

 شرط لتصفٌة نتائج Where و استخدمنا فً قسم Customer فً الجدول ID بالحقل Wadaeaa فً الجدول CustomerIDربط الحقل 

  لتحدٌد الحقول التً نرٌد إظهارها كنتائج للاستعلامSelectالاستعلام بحٌث نحصل على الودائع التً تبدأ بتارٌخ معٌن ثم نستخدم عبارة 

 لشبكة البٌانات الموجودة على ذلك DataSource قمنا بإنشاء متؽٌر من نوع تلك النافذة ثم ضبطنا قٌمة Form2ومن أجل إظهار النتائج فً 

 لتحوٌل النتائج إلى شكل ٌمكن إظهاره فً شبكة البٌانات ومن أجل ToListالنموذج إلى النتٌجة المعادة من الاستعلام مستخدمٌن الطرٌقة 

 بحٌث ٌصبح الكود كما Whereالحصول على الودائع التً تنتهً بتارٌخ معٌن ٌمكننا استخدام نفس الكود السابق بعد تعدٌل بسٌط فً قسم 

 ٌلً

    Private Sub Button2_Click() Handles Button2.Click 

        Dim d As Date = CDate(InputBox("Enter Date")) 

        Dim Dawa = From a In MyDataSet.Wadaeaa _ 

                  Join B In MyDataSet.Customers On _ 

                  a.CustomerID Equals B.ID _ 

                  Where DateAdd(DateInterval.Month, a.WadeeaPeriod, a.StartDate) = d _ 

                  Select B.CustomerName, B.CurrentAccountNumber, a.WadeaaNumber, _ 

                  a.WadeaaAmount, a.WadeeaPeriod, a.StartDate, _ 

                  EndDate = DateAdd(DateInterval.Month, a.WadeeaPeriod, a.StartDate) 

 

        Dim c As New Form2 

        c.DataGridView1.DataSource = Dawa.ToList 

        c.DataGridView1.Update() 

        c.ShowDialog() 

 

    End Sub 

 هل ٌمكنك شرح عمل الإجراء الثانً بنفسك ؟؟؟؟.  بٌن الإجراءٌن الأخٌرٌنWhereلاحظ الاختلاؾ فً قسم 

 Linq to XMLيقذية في  
 

Linq to Xml هً واجهة برمجة xml فً الذاكرة تدعم Linq تمكنك من العمل مع بٌانات xml المختلفة من داخل لؽة برمجة الـ .net 

Framework وهً مشابهة لـ Document Object Model  واختصارا DOM التً تضع الـ xml فً الذاكرة حٌث ٌمكنك الاستعلام من 

 فً أنها تزود نموذج ؼرضً DOM عن Linq to xmlالوثٌقة أو التعدٌل علٌها ثم ٌمكنك حفظها أو إرسالها بعد التعدٌل ولكن تختلؾ 

Object Model 2008 أخؾ وأسهل عند العمل علٌه وهً تستفٌد من تطوٌرات اللؽة فً الـ 

 فً الذاكرة للحصول xml الذي ٌمكنك من كتابة استعلامات من وثٌقة Linq هً التكامل مع Linq to Xmlوتكمن المٌزة الأهم التً تقدمها 

 وتقدم لك مٌزات إضافٌة مثل اكتشاؾ الأخطاء فً وقت الترجمة Xquery و xPathعلى مجموعة من العناصر والصفات التً تمتد لتشمل 

 XAttribute أو XElement إضافة إلى ترمٌز أقوى وإمكانٌة استخدام نتائج الاستعلامات كوسائط لبانٌات Debuggerودعم أفضل للمدقق 

 xml التً تمكن المطورٌن بسهولة من تحوٌل أشجار Functional Construction وهً تدعى xmlتوفر طرٌقة سهلة لإنشاء أشجار 

 . بسهولة من شكل إلى آخر

  ٌمثل طلب مشترٌات كالتالxmlً  فقد ٌكون لدٌك ملؾ xml من كتابة استعلامات من Linq to xml فً Linqوتمكنك إمكانٌات 

PurchaseOrder.xml 
<?xml version="1.0"?> 

<PurchaseOrder PurchaseOrderNumber="99503" OrderDate="1999-10-20"> 

  <Address Type="Shipping"> 

    <Name>Ellen Adams</Name> 

    <Street>123 Maple Street</Street> 

    <City>Mill Valley</City> 

    <State>CA</State> 

    <Zip>10999</Zip> 

    <Country>USA</Country> 

  </Address> 

  <Address Type="Billing"> 

    <Name>Tai Yee</Name> 

    <Street>8 Oak Avenue</Street> 

    <City>Old Town</City> 

    <State>PA</State> 

    <Zip>95819</Zip> 

    <Country>USA</Country> 

  </Address> 

  <DeliveryNotes>Please leave packages in shed by driveway.</DeliveryNotes> 

  <Items> 

    <Item PartNumber="872-AA"> 

      <ProductName>Lawnmower</ProductName> 

      <Quantity>1</Quantity> 

      <USPrice>148.95</USPrice> 

      <Comment>Confirm this is electric</Comment> 

    </Item> 

    <Item PartNumber="926-AA"> 

      <ProductName>Baby Monitor</ProductName> 

      <Quantity>2</Quantity> 

      <USPrice>39.98</USPrice> 

      <ShipDate>1999-05-21</ShipDate> 

    </Item> 

  </Items> 

</PurchaseOrder> 

 
  من أجل كل عنصر فً طلب المشترٌاتPartNumber ٌمكنك تشؽٌل استعلام للحصول على القٌمة المقابلة للصفة Linq to xmlفباستخدام 

Dim purchaseOrder As XElement = XElement.Load("PurchaseOrder.xml", LoadOptions.SetBaseUri 

Or LoadOptions.SetLineInfo) 

 

Dim partNos = _ 

    From item In purchaseOrder...<Item> _ 

    Select item.@PartNumber 

 
 وللحصول على هذه المعلومات 100 بالعناصر التً تحمل القٌمة أكثر من PartNumberوقد ترٌد الحصول على قائمة مرتبة باستخدام 

 ٌمكننا كتابة الاستعلام

Dim partNos = _ 

    From item In purchaseOrder...<Item> _ 

    Where (item.<Quantity>.Value * _ 

           item.<USPrice>.Value) > 100 _ 

    Order By item.<PartNumber>.Value _ 

    Select item 

 

 من xml ٌمكنك عمل العدٌد من الأشٌاء كتحمٌل ملؾ من القرص أو تخزٌن ملؾ إلى القرص أو إنشاء بٌانات Linq to xmlوباستخدام 

 xml من حٌث الإضافة والحذؾ والتعدٌل والتؤكد من صحة أشجار xml أو حتى التعامل مع أشجار Xpathالصفر أو الاستعلام باستخدام 

  من شكل إلى آخرxml أو استخدام مجموعة مما ورد هنا لتحوٌل أشجار XSDباستخدام 

 لإنشاء الشجرة وكلتا Linq APIs مباشرة فً الكود أو باستخدام xml إما بتعرٌؾ Visual Basic فً xmlوهناك طرٌقتان لإنشاء أشجار 

 xml شجرٌة كاملة فالكود التالً مثلا ٌنشئ عنصر xmlالطرٌقتٌن تمكنان الكود من عكس بنٌة 

Dim contact1 As XElement = _ 

    <contact> 

      <name>Patrick Hines</name> 

      <phone type="home">206-555-0144</phone> 

      <phone type="work">425-555-0145</phone> 

    </contact> 

 

 عن طرٌق تحدٌد اسم عنصر xml والتً تمكنك من الوصول إلى عناصر وصفات xmlوٌقدم فٌجول باٌزٌك عدة خصائص للتنقل عبر بنٌة 

xml الابن أو ٌمكنك استدعاء طرائق Linq لتحدٌد العناصر الأبناء لعنصر xml فالكود التالً مثلا ٌستخدم خصائص xml للإشارة إلى 

 xml للحصول على العناصر الأبناء وإخراجهم كعنصر Linq مستخدما استعلام xmlالصفات والعناصر الأبناء لعنصر 

' Place Imports statements at the top of your program.   

Imports <xmlns:ns="http://SomeNamespace"> 

 

Module Sample1 

 

    Sub SampleTransform() 

 

        ' Create test by using a global XML namespace prefix.  

 

        Dim contact = _ 

            <ns:contact> 

                <ns:name>Patrick Hines</ns:name> 

                <ns:phone ns:type="home">206-555-0144</ns:phone> 

                <ns:phone ns:type="work">425-555-0145</ns:phone> 

            </ns:contact> 

 

        Dim phoneTypes = _ 

          <phoneTypes> 

              <%= From phone In contact.<ns:phone> _ 

                  Select <type><%= phone.@ns:type %></type> _ 

              %> 

          </phoneTypes> 

 

        Console.WriteLine(phoneTypes) 

    End Sub 

 

End Module 

 
 كما فً الكود التالً الذي ٌرٌنا كٌؾ Imports باستخدام عبارة Xml لمجال أسماء Alias من تحدٌد اسم مستعار Visual Basicوٌمكنك 

  XML لاستٌراد مجال أسماء Importsٌمكننا استخدام العبارة 

Imports <xmlns:ns="http://someNamespace"> 

 

 وٌمكننا الحصول xml من أجل وثٌقة وعناصر xml ولتحدٌد محارؾ xmlحٌث ٌمكنك استخدام هذا الاسم المستعار للوصول إلى خصائص 

  كما فً المثالGetXmlNamespace من أجل أي بادئة مجال أسماء باستخدام المعامل XNamespaceعلى ؼرض 

' Place Imports statements at the top of your program.   

Imports <xmlns:ns="http://SomeNamespace"> 

 

Module GetXmlNamespaceSample 

    Sub RunSample() 

 

        ' Create test by using a global XML namespace prefix.  

 

        Dim contact = _ 

            <ns:contact> 

                <ns:name>Patrick Hines</ns:name> 

                <ns:phone ns:type="home">206-555-0144</ns:phone> 

                <ns:phone ns:type="work">425-555-0145</ns:phone> 

            </ns:contact> 

 

        ShowName(contact.<ns:phone>(0)) 

    End Sub 

 

    Sub ShowName(ByVal phone As XElement) 

        Dim qualifiedName = GetXmlNamespace(ns) + "contact" 

        Dim contact = phone.Ancestors(qualifiedName)(0) 

        Console.WriteLine("Name: " & contact.<ns:name>.Value) 

    End Sub 

 

End Module 

 

 ns باستخدام مجال الأسماء العام XElemntوٌرٌنا المثال التالً كٌفٌة إنشاء 

Dim contact1 As XElement = _ 

    <ns:contact> 

        <ns:name>Patrick Hines</ns:name> 

        <ns:phone type="home">206-555-0144</ns:phone> 

        <ns:phone type="work">425-555-0145</ns:phone> 

    </ns:contact> 

 

Console.WriteLine(contact1) 

 

 إلى الكود المكافئ الذي ٌستخدم Xml التً تحتوي الأسماء المستعارة لمجالات أسماء xmlوٌقوم مترجم فٌجول باٌزٌك بترجمة محارؾ 

 عند الترجمة والكود السابق ٌولد نفس الكود التنفٌذي الذي ٌولده الكود xmlns المستخدم فً تلك المجالات وباستخدام الصفة Xmlتدوٌن 

 التالً

Dim contact2 As XElement = _ 

    <ns1:contact xmlns:ns1="http://someNamespace"> 

        <ns1:name>Patrick Hines</ns1:name> 

        <ns1:phone type="home">206-555-0144</ns1:phone> 

        <ns1:phone type="work">425-555-0145</ns1:phone> 

    </ns1:contact> 

 

Console.WriteLine(contact2) 

 

 
  كما فً المثال التالXmlً العامة مع خصائص Xmlٌمكن استخدام مجالات أسماء 

Console.WriteLine("Contact name is: " & contact1.<ns:name>.Value) 

 Linq TO XMLبعض اسحخذايبت 

 تحتوي معلومات عن العملٌات xml فً فٌجول باٌزك انظر الكود التالً الذي ٌقوم بإنشاء وثٌقة xml لإنشاء وثائق Linqٌمكننا استخدام 

 الجارٌة فً النظام

Dim xmlProc = <MyProcesses> 

                  <%= From proc In System.Diagnostics.Process.GetProcesses() _ 

                      Select <process id=<%= proc.Id %>> 

                                 <name><%= proc.ProcessName %></name> 

                                 <threads><%= proc.Threads.Count %></threads> 

                             </process> %> 

              </MyProcesses> 

 

My.Computer.FileSystem.WriteAllText("d:\temp\processes.xml", xmlProc.ToString, False) 

Process.Start("d:\temp\processes.xml") 

 

 أنشؤنا العقد الفرعٌة وأدخلنا بقٌة الاستعلام لٌقوم select وأدخلنا فٌها بداٌة الاستعلام فٌها ثم فً قسم MyProcessesحٌث أنشؤنا العقدة 

 بسهولة حٌث نستخدم الكود التالً xmlكما ٌمكننا فٌجول باٌزٌك من الحصول على معلومات عن العقد فً وثٌقة . بضبط قٌم تلك العقد

  ListBoxلإظهار معلومات من الوثٌقة التً قمنا بإنشائها سابقا فً 

Dim xmlprocs1 = xmlProc...<process> 

For Each a In xmlprocs1 

    Me.ListBox1.Items.Add(a.<name>.Value & " " & a.@id) 

Next 

  كما فً الكودxml للاستعلام من وثٌقة Linqأو ٌمكننا  الحصول على نفس النتٌجة السابقة باستخدام استعلامات 

Dim xmlprocs1 = From pr In xmlProc...<process> _ 

                Select pr.<name>.Value, pr.@id 

 

For Each a In xmlprocs1 

    Me.ListBox1.Items.Add(a.name. & " " & a.id) 

Next 

  حساس xml كنتٌجة لاستعلام من وثٌقة موجودة سابقا على القرص وٌجب الانتباه إلى أن الاستعلام من وثائق xmlوقد نرٌد إنشاء وثٌقة 

 لحالة الأحرؾ

Dim myCusts = XDocument.Load("c:\MyCustomers.xml") 

 

Dim ukCustomers = <ukCustomers> 

                      <%= From cust In myCusts...<Customer> _ 

                          Where cust.<Country>.Value = "UK" _ 

                          Select cust %> 

                      </ukCustomers> 

  كما ٌلxmlًأو ٌمكننا كتابة استعلام مباشر من وثٌقة 

Dim xmlPlant = XDocument.Load(CurDir() & "\plants.xml") 

 

Dim qa = From p In xmlPlant...<PLANT> _ 

         Select name = p.<COMMON>.Value _ 

         Order By name 

 5وقد نرٌد كتابة استعلام من ذلك الملؾ لنحصل على النباتات التً تكلؾ أكثر من 

Dim qb = From p In xmlPlant...<PLANT> _ 

         Where CInt(p.<PRICE>.Value) > 5 _ 

         Select name = p.<COMMON>.Value 

 

 وادخل فً الثلاث خلاٌا الأولى من السطر الأول الكلمات التالٌة Excel لذا افتح Linq To xmlوقد نرٌد إنشاء وثٌقة إكسل من استعلام 

 ثم اجعل الخط سمٌكا ثم أدخل تحتها سطرا من البٌانات التجرٌبة كً نستخدمه فً تحدٌد القسم Country و Phone و Nameبالتسلسل 

 وانسخ Notepad ثم قم بفتح الملؾ الذي أنشؤته للتو بواسطة XML SpreadSheetالذي سنضع فٌه بٌاناتنا لاحقا ثم احفظ الملؾ بصٌؽة 

 ثم ألصق بعدها محتوٌات الحافظة فٌصبح لدٌك شٌئا شبٌها = Dim Sheetجمٌع محتوٌاته ثم عد إلى محرر الكود فً بٌئة التطوٌر واكتب 

 بالتالً

Dim Sheet = <?xml version="1.0"?> 

            <?mso-application progid="Excel.Sheet"?> 

            <Workbook xmlns="urn:schemas-microsoft-com:office:spreadsheet" 

                xmlns:o="urn:schemas-microsoft-com:office:office" 

                xmlns:x="urn:schemas-microsoft-com:office:excel" 

                xmlns:ss="urn:schemas-microsoft-com:office:spreadsheet" 

                xmlns:html="http://www.w3.org/TR/REC-html40"> 

                <DocumentProperties xmlns="urn:schemas-microsoft-com:office:office"> 

                    <Author>SamerSelo</Author> 

                    <LastAuthor>SamerSelo</LastAuthor> 

                    <Created>2008-09-19T20:31:43Z</Created> 

                    <Version>12.00</Version> 

                </DocumentProperties> 

                <ExcelWorkbook xmlns="urn:schemas-microsoft-com:office:excel"> 

                    <WindowHeight>7140</WindowHeight> 

                    <WindowWidth>15255</WindowWidth> 

                    <WindowTopX>120</WindowTopX> 

                    <WindowTopY>150</WindowTopY> 

                    <ProtectStructure>False</ProtectStructure> 

                    <ProtectWindows>False</ProtectWindows> 

                </ExcelWorkbook> 

                <Styles> 

                    <Style ss:ID="Default" ss:Name="Normal"> 

                        <Alignment ss:Vertical="Bottom"/> 

                        <Borders/> 

                        <Font ss:FontName="Arial" x:CharSet="178" x:Family="Swiss" ss:Size="11" 

                            ss:Color="#000000"/> 

                        <Interior/> 

                        <NumberFormat/> 

                        <Protection/> 

                    </Style> 

                </Styles> 

                <Worksheet ss:Name="1ورقة" ss:RightToLeft="1"> 

                    <Table ss:ExpandedColumnCount="3" ss:ExpandedRowCount="2" x:FullColumns="1" 

                        x:FullRows="1" ss:DefaultColumnWidth="54" ss:DefaultRowHeight="14.25"> 

                        <Row> 

                            <Cell><Data ss:Type="String">Name</Data></Cell> 

                        <Cell><Data ss:Type="String">Phone</Data></Cell> 

                             <Cell><Data ss:Type="String">Country</Data></Cell> 

                        </Row> 

                         <Row> 

                         <Cell><Data ss:Type="String">Test</Data></Cell> 

                              <Cell><Data ss:Type="Number">123456</Data></Cell> 

                         <Cell><Data ss:Type="String">Syr</Data></Cell> 

                         </Row> 

                    </Table> 

                    <WorksheetOptions xmlns="urn:schemas-microsoft-com:office:excel"> 

                        <PageSetup> 

                            <Header x:Margin="0.3"/> 

                            <Footer x:Margin="0.3"/> 

                            <PageMargins x:Bottom="0.75" x:Left="0.7" x:Right="0.7" x:Top="0.75"/> 

                        </PageSetup> 

                        <Selected/> 

                        <DisplayRightToLeft/> 

                        <Panes> 

                            <Pane> 

                                <Number>3</Number> 

                                <ActiveRow>1</ActiveRow> 

                                <ActiveCol>2</ActiveCol> 

                             </Pane> 

                        </Panes> 

                        <ProtectObjects>False</ProtectObjects> 

                        <ProtectScenarios>False</ProtectScenarios> 

                    </WorksheetOptions> 

                </Worksheet> 

                <Worksheet ss:Name="2ورقة" ss:RightToLeft="1"> 

                    <Table ss:ExpandedColumnCount="1" ss:ExpandedRowCount="1" x:FullColumns="1" 

                        x:FullRows="1" ss:DefaultColumnWidth="54" ss:DefaultRowHeight="14.25"> 

                    </Table> 

                    <WorksheetOptions xmlns="urn:schemas-microsoft-com:office:excel"> 

                        <PageSetup> 

                            <Header x:Margin="0.3"/> 

                            <Footer x:Margin="0.3"/> 

                            <PageMargins x:Bottom="0.75" x:Left="0.7" x:Right="0.7" x:Top="0.75"/> 

                        </PageSetup> 

                        <DisplayRightToLeft/> 

                        <ProtectObjects>False</ProtectObjects> 

                        <ProtectScenarios>False</ProtectScenarios> 

                    </WorksheetOptions> 

                </Worksheet> 

                <Worksheet ss:Name="3ورقة" ss:RightToLeft="1"> 

                    <Table ss:ExpandedColumnCount="1" ss:ExpandedRowCount="1" x:FullColumns="1" 

                        x:FullRows="1" ss:DefaultColumnWidth="54" ss:DefaultRowHeight="14.25"> 

                    </Table> 

                    <WorksheetOptions xmlns="urn:schemas-microsoft-com:office:excel"> 

                        <PageSetup> 

                            <Header x:Margin="0.3"/> 

                            <Footer x:Margin="0.3"/> 

                            <PageMargins x:Bottom="0.75" x:Left="0.7" x:Right="0.7" x:Top="0.75"/> 

                        </PageSetup> 

                        <DisplayRightToLeft/> 

                        <ProtectObjects>False</ProtectObjects> 

                        <ProtectScenarios>False</ProtectScenarios> 

                    </WorksheetOptions> 

                </Worksheet> 

            </Workbook> 

 

 فً بداٌتها وسنحتاج لاستٌرادها فً بداٌة كودنا xml التً ألصقناها للتو وجود بعض مجالات الأسماء الخاصة بـ xmlنلاحظ فً بداٌة وثٌقة 

 لذا فً قسم الاستٌرادات فً بداٌة الملؾ أدخل الاستٌرادات التالٌة حتى تساعدنا فً معرفة أسماء العناصر عند كتابة الاستعلام

Imports <xmlns="urn:schemas-microsoft-com:office:spreadsheet"> 

Imports <xmlns:o="urn:schemas-microsoft-com:office:office"> 

Imports <xmlns:x="urn:schemas-microsoft-com:office:excel"> 

Imports <xmlns:ss="urn:schemas-microsoft-com:office:spreadsheet"> 

 

 من محرر الكود ابحث عن الجزء الذي ٌمثل البٌانات التجرٌبٌة ثم قم بقصه وفً حالة مثالً سٌكون

<Row> 

    <Cell><Data ss:Type="String">Test</Data></Cell> 

    <Cell><Data ss:Type="Number">123456</Data></Cell> 

    <Cell><Data ss:Type="String">Syr</Data></Cell> 

</Row> 

  الذي عرفناه سابقا حٌث سنستخدم فٌه قطعة النص المقصوصة لتشكٌل شكل ناتج الاستعلامsheetأدخل الآن الاستعلام التالً قبل المتؽٌر 

Dim Customers = From Customer In db.Customers _ 

                Order By Customer.CompanyName _ 

                Select <Row> 

                           <Cell><Data ss:Type="String"><%= Customer.CompanyName %></Data></Cell> 

                           <Cell><Data ss:Type=" String"><%= Customer.Phone %></Data></Cell> 
                           <Cell><Data ss:Type="String"><%= Customer.Country %></Data></Cell> 

                       </Row> 

  سابقا وقم بإدخال السطر التالً فً نفس المكانxmlثم انتقل للمكان الذي قصصنا منه قطعة 

<%= Customers %> 

 انتقل إلى الأعلى قلٌلا وعدل السطر

<Table ss:ExpandedColumnCount="3" ss:ExpandedRowCount="2" x:FullColumns="1" 

 إلى

<Table ss:ExpandedColumnCount="3" ss:ExpandedRowCount=<%= Customers.Count + 1 %> x:FullColumns="1" 

 وكل ما تبقى هو الحفظ وعرض الملؾ وهذا ٌتم بالكود التالً

Sheet.Save("d:\temp\customers11.xml") 

Process.Start("d:\temp\customers11.xml") 

 O/R Designer و Linq to SQLجعرف عهى 

 

 للتعامل المباشر مع قاعدة بٌانات Entity Classes لإنشاء O/R Designerسنقوم هنا بإنشاء برنامج بسٌط ٌعرض لنا كٌفٌة استخدام 

 الشهٌرة التً ٌمكنك الحصول علٌها بسهولة من موقع NorthWind وسؤعتمد حالٌا على قاعدة بٌانات SQL Serverموجودة فً 

 حتى ٌسهل علٌنا Server Explorerماٌكروسوفت وتثبٌتها لدٌك كما ٌتوجب علٌك إنشاء اتصال لتلك القاعدة من داخل بٌئة التطوٌر فً نافذة 

العمل على كل حال موضوع تركٌب قاعدة البٌانات وإنشاء الاتصال خارج عن مجال دورتنا وأفترض أنه لدٌك بعض الأساسٌات التً 

 وبعض فٌدٌوهات ماٌكروسوفت التعلٌمٌة إن احتجت لمساعدة فً هذه MSDNتساعدك فً التعامل مع هكذا أمور وٌمكنك مراجعة مكتبة 

 الأمور

 Linq To SQL ومن  صندوق الحوار اختر Add New انقر بزر الفؤرة الٌمٌنً على مشروعك ومن القائمة اختر Solution Explorerمن 

Classes ثم قم بتسمٌة الفئة الجدٌدة NorthWind.dbml ثم اضؽط Add فٌفتح لك واجهة O/R Designer الفارؼة وإن كانت مؽلقة 

  من مستكشؾ الحل لفتحهاNorthWind.dbmlٌمكنك النقر المزدوج على 

 ثم O/R Designer إلى واجهة Customers واسحب الجدول Server Explorer من Northwindوسع الاتصال الخاص بقاعدة البٌانات 

 فٌفتح لك المعالج المؤلوؾ وفً حالتنا هذه سنختار Add New Data Source اضؽط Data Sourcesقم بالحفظ وبعدها انتقل إلى نافذة 

Object بما أنه المناسب لعملنا هنا ثم نضؽط Next ثم وسع العقد فً نافذة المعالج واختر Customer ثم اضؽط Next ثم Finish 

 إلى سطح النافذة فٌتم إنشاء شرٌط أدوات و DataSources من واجهة Customerعد إلى محرر النماذج وقم بسحب الجدول 

DataGridView من أجلك افتح السهم الصؽٌر أعلى ٌمٌن DataGridView واضبط الخٌار Dock in parent Container لجعل شبكة 

البٌانات تملأ كافة المساحة الفارؼة فً النموذج وكما تلاحظ هنا أن بٌئة التطوٌر تسهل علٌنا الكثٌر من الأمور هنا من إنشاء للتحكمات 

 والفئات والربط بٌنها مما ٌوفر علٌنا الكثٌر من العمل

 Loadمن أجل إظهار البٌانات على النموذج سنحتاج لكتابة بعض الكود لذا انتقل لمحرر الكود الخاص بالنموذج وأنشئ معالج للحدث 

 Db الخاص بالنموذج أدخل المتؽٌر التالً بحٌث ٌكون عاما على مستوى النموذج حٌث أن المتؽٌر Loadللنموذج  وقبل بداٌة تعرٌؾ الحدث 

 Linq والتً تشكل اتصالنا الفعلً مع قاعدة البٌانات بما أنها نقطة الدخول الرئٌسٌة بالنسبة لـ NorthWindDataContextهنا هو كٌان من 

To SQL 

Private Db As New NorthWindDataContext 

  ٌزودنا بالبٌانات التً سٌتم إظهارهاLinq للنموذج سنضع استعلام Loadفً الحدث 

Dim AllCustomers = From cust In Db.Customers _ 

                   Order By cust.CustomerID _ 

                   Select cust 

 قمنا باستخدامه منذ بداٌة الدورة حتى الآن كل ما علٌنا لإظهار البٌانات هو ضبط قٌمة الخاصٌة Linq عادي كؤي استعلام Linqهذا استعلام 

DataSource لـ CustomerBindingSourec إلى استعلامنا AllCustomers أدخل السطر التالً مباشرة بعد الاستعلام ثم شؽل 

 البرنامج وتؤكد من ظهور البٌانات

Me.CustomerBindingSource.DataSource = AllCustomers 

من أجل حفظ التعدٌلات التً ربما سنقوم بها إلى قاعدة البٌانات عد إلى محرر النماذج واجعل زر الحفظ الموجود على شرٌط الأدوات 

Enabled ثم انقر علٌه نقرا مزدوجا لإنشاء معالج لحدث النقر علٌه والانتقال لمحرر الكود وأدخل الكود التالً الذي سٌقوم بحفظ البٌانات من 

 أجلنا

Me.Validate() 

Me.CustomerBindingSource.EndEdit() 

 

Try 

    Db.SubmitChanges() 

    MsgBox("Changes Saved") 

Catch ex As Exception 

    MsgBox(ex.Message) 

End Try 

 العائدة لـ EndEdit أولا لجعل جمٌع التحكمات على النموذج أن تتحقق من قٌمها ثم استدعٌنا الطرٌقة Me.Validateحٌث استخدمنا 

CustomerBindingSource من أجل التؤكد من أن جمٌع عملٌات التحرٌر على البٌانات قد تم إنهاإها ثم استخدمنا الطرٌقة 

SubmitChanges العائدة لـ DataContext التً ستقوم بحفظ البٌانات فعلٌا إلى قاعدة البٌانات واستخدمنا بلوك Try … Catch من أجل 

 هنا بدون محددات إلا أنه ٌمكن استخدامها SubmitChangesالتقاط أي خطؤ ربما نصادفه أثناء عملٌة الحفظ ومع أننا استخدمنا الطرٌقة 

 التً توقؾ عملٌة تحدٌث البٌانات FailOnFirstConflict الذي ٌمتلك إحدى قٌمٌتٌن ConflictModeبتمرٌر محدد وحٌد من نوع التعداد 

 وهً تتابع عملٌة الحفظ حتى لو حدثت ContinueOnConflictإلى قاعدة البٌانات عند حصول أول تضارب وهً القٌمة الافتراضٌة و 

 تضاربات وتقوم بتجمٌع هذه التضاربات وتعٌدها بعد انتهاء عملٌة تحدٌث البٌانات إلى قاعدة البٌانات 

 ٌمكنك عند هذه النقطة تشؽٌل البرنامج والتؤكد من أنك تستطٌع إضافة وتعدٌل وحذؾ البٌانات بدون أٌة مشاكل

نرٌد الآن إضافة بعض التصفٌة على البٌانات فً النموذج ولهذا الؽرض قم بإضافة صندوق نصوص وزر أوامر إلى شرٌط الأدوات فً 

أعلى النموذج وانقر نقرا مزدوجا على الزر حتى ننتقل لمحرر الكود وفً إجراء معالجة حدث النقر علٌه أدخل الآن الاستعلام التالً الذي 

  إلى الاستعلام الجدٌد لإظهار النتائجCustomerBindingSource لـ DataSourceأتبعناه بكود تحدٌث الخاصٌة 

Dim CustNameQuery = From cust In Db.Customers _ 

                    Where cust.ContactName Like _ 

                    Me.ToolStripTextBox1.Text & "*" _ 

                    Order By cust.CustomerID _ 

                    Select cust 

 

Me.CustomerBindingSource.DataSource = CustNameQuery 

 

أنا لا أقوم بشرح الاستعلامات هنا بافتراض أنك متابع ممتاز معً منذ البداٌة وأصبحت متآلفا مع صٌؽة وطرٌقة كتابة هذه : ملاحظة

 الاستعلامات كما ٌمكنك كتابة أٌة استعلامات ترٌدها هنا وبؤي شكل هنا تماما كما فعلنا فً الدروس السابقة

Linq To Sql Master/Detail 

 

 NorthWind.dbml إلٌه تماما كما فعلنا فً الدرس السابق وسمها Linq To SQL Classesافتح مشروع فٌجول باٌزٌك جدٌد وقم بإضافة 

 وقم بسحبهما Customers و Orders وقم باختٌار الجدولٌن NorthWind وسع عقدة اتصال قاعدة البٌانات Server Explorerثم من 

 ونرى أن بٌئة التطوٌر قامت Order و الثانٌة Customer الأمر الذي ٌنشئ فئتٌن من أجلنا الأولى تدعى O/R Designerمعا إلى نافذة 

بضبط العلاقة بٌنهما تلقائٌا كما نلاحظ أن جمٌع خصائص كلتا الفئتٌن تماثل تماما الحقول الموجودة فً الجدولٌن الموجودٌن فً قاعدة 

 البٌانات قم بحفظ المشروع الآن قبل المتابعة

 واختر من الصفحة الأولى فً المعالج أن النوع الذي نرٌد الاتصال به Add New Data Source واختر Data Sourcesانتقل إلى نافذة 

 اختٌار الجدول  بدون فقط Customer للانتقال للصفحة التالٌة من المعالج  ثم وسع العقد وقم باختٌار الجدول Next ثم اضؽط Objectهو 

Order  ثم اضؽطNext ثم Finish وستلاحظ فً نافذة Data Sources أنه قد تم إدراج الفئة Customer وتظهر الفئة Orders ككائن 

  فقط فً المعالجCustomerفرعً منها ولهذا قمنا باختٌار 

 فً أعلى النافذة ثم قم BindingNavigator على النافذة وشرٌط أدوات DataGridView إلى سطح النافذة لٌتم إنشاء Customerاسحب 

  أخرى أسفل الأولىDataGridView إلى سطح النافذة لإنشاء Ordersبسحب 

 للنموذج وقبل تعرٌؾ الإجراء مباشرة قم بإدخال المتؽٌر Loadانتقل إلى محرر الكود الخاص بالنموذج وقم بإنشاء إجراء لمعالجة حدث 

 نقطة أنها بما البٌانات قاعدة مع الفعلً اتصالنا تشكل والتً NorthWindDataContextالعام التالً على مستوى النموذج لٌكون كٌانا من 

 Linq To SQL لـ بالنسبة الرئٌسٌة الدخول

Private Db As New NorthWindDataContext 

  للنموذج وقم بتجربة البرنامجLoadولجعل البٌانات تظهر فً كلتا الشبكتٌن أدخل الكود التالً فً إجراء الحدث 

Me.CustomerBindingSource.DataSource = Db.Customers 

 فً الخلفٌة بسبب العلاقة الموجودة بٌن Linq to Sql قد تم جلبها وهذا تقوم به من أجلنا Ordersوعند تجربة البرنامج ستجد أن بٌانات 

 والذي ٌحدث فعلٌا هو أنك عندما تختار سجلا من O/R Designer فً بداٌة العمل بواسطة Entity Classesالجدولٌن عندما أنشؤنا 

Customers ٌتولد تلقائٌا استعلام ٌجلب بٌانات Orders المرتبطة بهذا السجل بموجب العلاقة بٌن الجدولٌن وٌظهرها فً شبكة البٌانات 

 الثانٌة

لكً نستطٌع حفظ أٌة تعدٌلات أو إضافات نقوم بها على قاعدة البٌانات عد إلى محرر النماذج واجعل زر الحفظ الموجود على الشرٌط فً 

 وانقر علٌه نقرا مزدوجا من أجل إنشاء إجراء معالجة لحدث النقر علٌه والانتقال لمحرر الكود وسٌكون كود الحفظ Enabledأعلى النموذج 

 شبٌها بالكود الذي استخدمناه فً الدرس السابق

Me.Validate() 

Me.OrdersBindingSource.EndEdit() 

Me.CustomerBindingSource.EndEdit() 

 

Try 

    Me.Db.SubmitChanges() 

    MsgBox("Changes Saved.") 

Catch ex As Exception 

    MsgBox(ex.Message) 

End Try 

 جرب البرنامج وتؤكد من أن جمٌع عملٌات الحذؾ والتعدٌل والإضافة تعمل

 لشرٌط الأدوات فً أعلى النموذج الذي سنقوم بملئه بؤسماء الدول المتوفرة حتى نستطٌع ComboBoxانتقل إلى محرر النماذج وأضؾ 

 للنموذج بالكود التالً الذي سٌقوم Loadاختٌار الزبائن الموجودٌن فً دولة معٌنة انتقل الآن لمحرر الكود واستبدل كامل محتوٌات الحدث 

 فً عبارة الاستعلام كً لا ٌجلب لنا نتائج مكررة عندما تكون هناك نتائج distinctبملئ صندوق القائمة بؤسماء الدول لاحظ استخدام 

 متشابهة معادة من قاعدة البٌانات

Dim CusCountry = From co In Db.Customers _ 

                 Where co.Country <> String.Empty _ 

                 Order By co.Country _ 

                 Select co.Country Distinct 

 

Me.ToolStripComboBox1.Items.Clear() 

 

For Each co In CusCountry 

    Me.ToolStripComboBox1.Items.Add(co) 

Next 

 وأدخل فٌه كود الاستعلام التالً لكً ٌتم إظهار ToolStripComboBox1 لـ SelectedIndexChangedأنشئ إجراء معالجة لحدث 

 الزبائن الموجودة فً دولة معٌنة عند اختٌارها من صندوق القائمة المركبة

Dim CustQuery = From co In Db.Customers _ 

                Where co.Country = Me.ToolStripComboBox1.SelectedItem.ToString _ 

                Select co 

 

Me.CustomerBindingSource.DataSource = CustQuery 

وللحصول على مجموع أجور الشحن لطلبات الزبون المحدد أضؾ صندوق نصوص أسفل النموذج ثم أنشئ إجراء لمعالجة حدث 

CurrentChanged الخاص بـ CustomerBindingSourceًوأدخل فٌه الكود التال  

Dim ro = CType(Me.CustomerBindingSource.Current, Customer) 

 

Dim FriSum = Aggregate ord In Db.Orders _ 

             Where ord.CustomerID = ro.CustomerID _ 

             Into Sum(ord.Freight) 

 

Me.TextBox1.Text = FriSum.ToString 

 لتحوٌل ناتج الخاصٌة CType وهنا اضطررنا لاستخدام CustomerBindingSourceحٌث حصلنا أولا على معلومات السجل الحالً فً 

Current النوع من قٌمة تعٌد أنها بما إلى النوع المطلوب Object  وبما أننا لم نحدد نوع المتؽٌرro عند التصرٌح عنه فستقوم بذلك نٌابة 

  لحساب مجموع أجور شحن الطلبات الخاصة بالزبون الحالً aggregateعنا مٌزة الاستدلال المحلً على النوع ثم استخدمنا استعلام 

أضؾ زرا وصندوق نصوص إلى النموذج حتى نستطٌع إظهار طلبات الزبون المحدد بعد تارٌخ معٌن ولكً لانفقد الارتباط بٌن شبكتً 

 وأدخل فٌه الكود التالً قبل الكود CustomerBindingSource الخاص بـ CurrentChangedالبٌانات انتقل إلى إجراء معالجة الحدث 

  بما أننا سنؽٌرها لاحقا عندما سننفذ استعلامناOrdersBindingSourceالموجود فٌه كً نعٌد ضبط القٌم الصحٌحة لـ 

Me.OrdersBindingSource.DataSource = CustomerBindingSource 

Me.OrdersBindingSource.DataMember = "Orders" 

Me.TextBox2.Text = String.Empty 

 أنشئ إجراء لمعالجة حدث النقر على الزر الذي أضفناه للنموذج وأدخل فٌه الكود التالً

Dim ro = CType(Me.CustomerBindingSource.Current, Customer) 

 

If IsDate(Me.TextBox2.Text) Then 

    Dim SpOrd = From ord In Db.Orders _ 

                Where ord.CustomerID = ro.CustomerID _ 

                And ord.OrderDate >= CDate(Me.TextBox2.Text) _ 

                Select ord 

 

    Me.OrdersBindingSource.DataSource = SpOrd 

Else 

    Me.OrdersBindingSource.DataSource = CustomerBindingSource 

    Me.OrdersBindingSource.DataMember = "Orders" 

    Me.TextBox2.Text = String.Empty 

End If 

 للتحقق CDate و الدالة If ثم استخدمنا عبارة ro فً المتؽٌر Customersحٌث حصلنا فً البداٌة على معلومات السجل الحالً بالنسبة لـ 

 OrdersBindingSourceمن أن المستخدم قد قام بإدخال تارٌخ صحٌح قبل المتابعة بالاستعلام وإن لم ٌكن تارٌخا صحٌحا نعٌد ضبط قٌم 

إلى القٌم الأصلٌة ونقوم بتفرٌػ النص الموجود فً صندوق النصوص وإن كان قد ادخل تارٌخا صحٌحا ننشئ استعلام ٌقوم بجلب الطلبات 

 فً الاستعلام ثم نقوم Whereالتً تارٌخها بعد التارٌخ الموجود فً صندوق النصوص والخاصة بالزبون الحالً كما هو ظاهر فً قسم 

 إلى استعلامنا كً ٌتم عرض البٌانات المعادة من الاستعلام فً شبكة البٌانات OrdersBindingSource لـ DataSourceبضبط الخاصٌة 

 الثانٌة

  يذويبLinq To SQLيثبل سريع عن كيفية إنشبء فئبت 

 

 بسهولة من Entity Classes أو كما تدعى أٌضا Linq To SQL ٌقوم بإنشاء فئات O/R Designerرأٌنا فً الدروس السابقة كٌؾ أن الـ 

 أجلنا ومع ذلك ٌمكننا القٌام بذلك ٌدوٌا وسؤقوم فً هذا الدرس باستعراض سرٌع لكٌفٌة عمل ذلك من أجل العلم بالشئ

 LinqConsoleApp وسمه Console Application النوع من جدٌدا مشروعا أنشئ

  Ok اضؽط ثم System.Data.Linq اختر net .صفحة ومن Add Reference اختر Project قائمة من •

 التالٌة الاستٌرادات أضؾ الكود محرر أعلى فً الملؾ بداٌة فً •
Imports System.Data.Linq 

Imports System.Data.Linq.Mapping 

 

 تعرٌؾ قبل Table الصفة نضٌؾ الفئة هذه ولإنشاء بٌانات قاعدة جدول وفق منظمة فئة عن عبارة عن والذي Entity Class الآن سنضٌؾ

 الجدول أعمدة لتمثل الخصائص المناسبة إضافة علٌنا سٌكون ثم البٌانات قاعدة فً الجدول اسم تحدد التً Name الخاصٌة تمتلك والت الفئة

 IsPrimaryKey مثل العمود لتعرٌؾ خصائص عدة تمتلك والتً Column الصفة باستخدام البٌانات قاعدة فً الأعمدة مع الربط تحدٌد وٌتم

 قٌمة سٌخزن الذي الخاص الحقل اسم تحدد التً Storage والخاصٌة لا أم أساسً مفتاح العمود كان إذا فٌما تحدد قٌمة منطقٌة تمتلك التً

 فً قاعدة البٌانات طبعا Customers والتً ستمثل الجدول Sub Main وقبل Module الـ تعرٌؾ بعد التالٌة الفئة أضؾ – الخاصٌة

عرفت بعض الخصائص من أجل بعض الحقول الموجودة فً الجدول هنا ولٌس جمٌع الحقول وذلك من اجل الاختصار هنا وإن قررت 

 استخدام هذه الطرٌقة عملٌا علٌك تعرٌؾ الفئة بحٌث تكون مطابقة تماما للجدول الذي تمثله

  <Table(Name:="Customers")> _ 

    Public Class Customer 

 

        Private _CustomerID As String 

        <Column(IsPrimaryKey:=True, Storage:="_CustomerID")> _ 

        Public Property CustomerID() As String 

            Get 

                Return Me._CustomerID 

            End Get 

            Set(ByVal value As String) 

                Me._CustomerID = value 

            End Set 

        End Property 

 

        Private _City As String 

        <Column(Storage:="_City")> _ 

        Public Property City() As String 

            Get 

                Return Me._City 

            End Get 

            Set(ByVal value As String) 

                Me._City = value 

            End Set 

        End Property 

 

       Private _ContactName As String 

        <Column(Storage:="_ContactName")> _ 

        Public Property ContactName() As String 

            Get 

                Return Me._ContactName 

            End Get 

            Set(ByVal value As String) 

                Me._ContactName = value 

            End Set 

        End Property 

 

    End Class 

 ٌعتبر والذي DataContext ؼرض تعرٌؾ إلى سنحتاج لذا سنستخدمها التً NorthWind البٌانات قاعدة مع اتصال إنشاء إلى سنحتاج

 وذلك باعتبار أن قاعدة Sub Main الإجراء فً التالً الكود سطر أدخل – البٌانات قاعدة فً وتخزٌنها البٌانات لقراءة الرئٌسٌة القناة

  لدي عند تجربة هذا المثالD:\TEMP الموجود فً المجلد NORTHWND.MDFالبٌانات موجودة فً الملؾ 

Dim db As New DataContext("D:\TEMP\NORTHWND.MDF") 

 الكود بعد مباشرة التالً الكود أدخل Customers الجدول من الاستعلام فً نستخدمه كً Table(of Customer)كائن  بإنشاء سنقوم ثم

 السابق

Dim Customers As Table(Of Customer) = _ 

        db.GetTable(Of Customer)() 

 

 البٌانات قاعدة على تنفٌذها سٌتم التً Sql أوامر الكونسول نافذة على سٌطبع الذي الكود من التالً السطر أضؾ

db.Log = Console.Out 

 السابق مباشرة الكود بعد الاستعلام التالً كود أدخل – لندن فً موجودٌن الزبائن من أي لٌستعلم Linq استعلام الآن سنكتب

Dim custQuery = From cust In Customers _ 

                Where cust.City = "London" _ 

                Select cust 

 

للدوران عبر نتائج الاستعلام كما رأٌنا فً الدروس السابقة ثم طباعتها على نافذة  For Each سنستخدم حلقة الاستعلام وللحصول على نتائج

 الكونسول 

 
 

Console.WriteLine("Number Of Records: " & custQuery.Count.ToString) 

 

For Each CustObj In custQuery 

    Console.WriteLine(CustObj.CustomerID.ToString & ", " & _ 

                      CustObj.City & ", " & CustObj.ContactName) 

Next 

 

Console.ReadLine() 

 

 

 وهذا هو الكود الكامل للمشروع
Imports System.Data.Linq 

Imports System.Data.Linq.Mapping 

 

Module Module1 

 

    <Table(Name:="Customers")> _ 

    Public Class Customer 

 

        Private _CustomerID As String 

        <Column(IsPrimaryKey:=True, Storage:="_CustomerID")> _ 

        Public Property CustomerID() As String 

            Get 

                Return Me._CustomerID 

            End Get 

            Set(ByVal value As String) 

                Me._CustomerID = value 

            End Set 

        End Property 

 

        Private _City As String 

        <Column(Storage:="_City")> _ 

        Public Property City() As String 

            Get 

                Return Me._City 

            End Get 

            Set(ByVal value As String) 

                Me._City = value 

            End Set 

        End Property 

 

        Private _ContactName As String 

        <Column(Storage:="_ContactName")> _ 

        Public Property ContactName() As String 

            Get 

                Return Me._ContactName 

            End Get 

            Set(ByVal value As String) 

                Me._ContactName = value 

            End Set 

        End Property 

 

    End Class 

 

    Sub Main() 

        Dim db As New DataContext("D:\TEMP\NORTHWND.MDF") 

 

        Dim Customers As Table(Of Customer) = _ 

                db.GetTable(Of Customer)() 

 

        db.Log = Console.Out 

 

        Dim custQuery = From cust In Customers _ 

                        Where cust.City = "London" _ 

                        Select cust 

 

        Console.WriteLine("Number Of Records: " & custQuery.Count.ToString) 

 

        For Each CustObj In custQuery 

            Console.WriteLine(CustObj.CustomerID.ToString & ", " & _ 

                              CustObj.City & ", " & CustObj.ContactName) 

        Next 

 

        Console.ReadLine() 

 

    End Sub 

 

End Module 

  Linqأيثهة عهى اسحعلايبت 

 

فً هذا الدرس سؤورد بعض الأمثلة على استعلامات مختلفة كً نتعرؾ أكثر على أسلوب كتابة هذه الاستعلامات ولن أتطرق إلى طرق 

 … Forاستخدام هذه الاستعلامات بما أننا رأٌنا كٌؾ ٌمكن عمل ذلك من خلال الأمثلة الواردة فً الدروس السابقة إما بالدوران باستخدام حلقة 

Each ًأو بالإسناد المباشر لمتؽٌر أو تحكم أو الإظهار ف DataGridView 

 مثال بسٌط على استخدام الدالات التجمٌعٌة فً الاستعلام للحصول على القٌمة العظمى لحقل ٌحتوي على مجموعة قٌم

Dim Wod = Aggregate Wdt In ads.WorkingDate _ 

          Into Mdt = Max(Wdt.WorkingDate) 

 

 مثال آخر على استخدام الدالات التجمٌعٌة للحصول على مجموع ناتج عملٌة طرح حقلٌن

Dim TempBal = Aggregate AccBa In AccMov _ 

              Into Bal = Sum(AccBa.DebitAmmount - AccBa.CreditAmmount) 

 

  فً الاستعلام للحصول على مجموعة نتائج محددةWhere فً قسم Likeمثال بسٌط آخر ٌستخدم 

Dim AccMov = From AccBal In Accds.AccountMovements _ 

             Where AccBal.AccountNumber Like (AccNum & "*") _ 

             Select AccBal 

 

  للحصول على نتائج الاستعلام بحٌث لا ٌتؤثر شرط التصفٌة بحالة الأحرؾWhere فً قسم ToLowerمثال ٌستخدم الدالة 

Dim Qts = From cu In Db.Customers _ 

          Where cu.Country.ToString.ToLower Like "po*".ToLower _ 

          Select cu 

 

  لترتٌب النتائج تصاعدٌاOrder Byمثال بسٌط آخر ٌستخدم 

Dim ParentList = From ParLst In AccountsDataSet.Accounts _ 

                 Where ParLst.IsChildAccount = False _ 

                 Order By ParLst.AccountNumber _ 

                 Select ParLst 

 Order By بعد حقل الفرز فً قسم Descendingومن أجل الترتٌب تنازلٌا ٌصبح الاستعلام كما ٌلً وذلك بإضافة 

Dim ParentList = From ParLst In AccountsDataSet.Accounts _ 

                 Where ParLst.IsChildAccount = False _ 

                 Order By ParLst.AccountNumber Descending _ 

                 Select ParLst 

 

 لربط جدولٌن للحصول على القٌم Joinهذا المثال ٌقوم بإعادة تارٌخ محتسب من تارٌخ وفترة زمنٌة مخزنٌن فً قاعدة بٌانات مع استخدام 

 منهما

Dim Edat = From de In DsDesposits.Deposits _ 

           Join Dp In DsDesposits.InterestRates _ 

           On de.InterestID Equals Dp.InterestID _ 

           Select EndDate = DateAdd(DateInterval.Month, _ 

                                   intrst.First, DepDet.First.StartDate) 

 لربط جدولٌن من أجل الاستعلام منهما ثم اختٌار حقول محددة كنتٌجة للاستعلام ثم استخدام الاستعلام Joinمثال آخر عن استعلام ٌستخدم 

 للاستعلام من نتٌجة ربط استعلام وجدول للحصول على النتائج Joinداخل استعلام آخر من أجل الحصول على نتٌجة إضافٌة ثم استخدام 

 المرؼوبة

Dim MovNam = From Acc In SampleDatabaseDataSet.ACCOUNTS _ 

             Join Mov In SampleDatabaseDataSet.ACCOUNTS_MOVEMENTS _ 

             On Acc.ACC_NUMBER Equals Mov.ACC_NUMBER _ 

             Select Mov.MOVMENT_DATE, Mov.ACC_NUMBER, Acc.ACC_NAME, _ 

             Mov.DEBIT_AMOUNT, Mov.CREDIT_AMOUNT 

 

Dim AccBal = From AcBa In MovNam _ 

             Group By acc_number = AcBa.ACC_NUMBER _ 

             Into Balance = Sum(AcBa.DEBIT_AMOUNT - AcBa.CREDIT_AMOUNT) 

 

 

Dim AccBalName = From ab In AccBal Join ac In SampleDatabaseDataSet.ACCOUNTS _ 

                 On ab.acc_number Equals ac.ACC_NUMBER _ 

                 Select ab.acc_number, ac.ACC_NAME, ab.Balance 

 

  لإنشاء استعلام ٌعطٌنا النتائج نتٌجة الاستعلام من أربعة جداول مختلفةJoinهذا المثال ٌستخدم 

Dim ViewData = From Cu In DsDesposits.Customers _ 

               Join Dp In DsDesposits.Deposits On Dp.CustomerID Equals Cu.CustomerID _ 

               Join DepPer In DsDesposits.InterestRates On DepPer.InterestID Equals Dp.InterestID _ 

               Join CaDp In DsDesposits.CalculatedDepsits On CaDp.DepositID Equals Dp.DepositID _ 

               Where CaDp.CalculationReason = 2 _ 

               And CaDp.CalculationDate >= FromDate _ 

               And CaDp.CalculationDate <= ToDate _ 

               Select Cu.Name, Dp.DepositNumber, Dp.DepositAmount, Dp.StartDate, _ 

               CaDp.CalculationDate, DepPer.DepositPreiod 

 

 للحصول على نتٌجة من قائمة قٌم فهً ؼٌر موجودة هنا لذا نقوم sql المستخدمة فً استعلامات  inهذا المثال فٌه نقطتٌن الأولى بخصوص 

 فً الاستعلام للحصول على نفس النتٌجة والثانٌة هً أننا نستطٌع استخدام نتٌجة استعلام Containsبوضع القٌم فً مصفوفة ثم نستخدم 

 كدخل لاستعلام آخر

Dim Vlu() As Byte = {0, 3, 4, Nothing, 6} 

Dim DepDet = From De In DsDesposits.Deposits _ 

             Where De.DepositID = Dr.DepositID _ 

             And Vlu.Contains(De.DepositStatus) 

 

Dim Intrst = From Irs In DsDesposits.InterestRates _ 

             Where Irs.InterestID = DepDet.First.InterestID _ 

             Select Irs.DepositPreiod, Irs.InterestRate 

 

  وطرٌقة المصفوفة معاJoinمثال آخر ٌستخدم 

Dim Vlu() As Byte = {0, Nothing, 3, 4, 6} 

mDep = From d In DsDesposits.Deposits _ 

       Join i In DsDesposits.InterestRates On d.InterestID Equals i.InterestID _ 

       Where DateAdd(DateInterval.Month, i.DepositPreiod, d.StartDate) <= Now.Date _ 

       And Vlu.Contains(d.DepositStatus) _ 

       Select d.DepositID 

 

 لاستثناء النتائج المعادة من Except فً نهاٌة الاستعلام الأول استخدمنا الطرٌقة selectهذا المثال ٌستخدم استعلامٌن متداخلٌن ففً قسم 

 Exceptالاستعلام الثانً الممرر كمحدد للطرٌقة 

Deps = (From d In DsDesposits.Deposits _ 

        Join i In DsDesposits.InterestRates _ 

        On i.InterestID Equals d.InterestID _ 

        Where (d.StartDate <= New Date(Ryear, 12, 31) _ 

        And vlu.Contains(d.DepositStatus)) _ 

        And DateAdd(DateInterval.Month, i.DepositPreiod, d.StartDate) >= New Date(Ryear, 12, 31) _ 

        Select d.DepositID).Except( _ 

        From ca In DsDesposits.CalculatedDepsits _ 

        Where ca.EndDate = New Date(Ryear, 12, 31) _ 

        Select ca.DepositID) 

 Compiled Queries الاستعلامات المترجمة

 

 الاستعلام مرة واحدة ثم Compileعندما ٌكون لدٌنا تطبٌق ٌستخدم استعلامات متشابهة العدٌد من المرات ٌمكنك زٌادة الأداء عبر ترجمة 

تنفٌذه العدٌد من المرات لاحقا عبر تمرٌر محددات مختلفة فً كل مرة فقد ٌكون لدٌك تطبٌق ٌقوم بالاستعلام عن جمٌع الزبائن الموجودٌن 

 استخدام الاستعلامات المترجمة لهذا Linq to Sqlفً مدٌنة محددة بحٌث ٌتم تمرٌر اسم المدٌنة فً وقت التنفٌذ من قبل المستخدم وهنا تدعم 

 الؽرض

 التً تزودنا بإمكانٌة الترجمة والتخزٌن المإقت للاستعلامات من اجل الاستخدام وهذه CompiledQueryحٌث توفر لنا الفرٌموورك الفئة 

 التً تعٌد الاستعلام كتعبٌر لمدا Expression وهً تمتلك خاصٌة وحٌدة هً System.Data.Linqالفئة متواجدة فً مجال الأسماء 

Lambda Expression وهً تمتلك بعض الطرق ولكن الطرٌقة الأهم والتً ٌتم استخدامها هنا هً الطرٌقة Compile التً تمتلك عدة 

 أشكال محملة متشابهة 

Public Shared Function Compile(Of TArg0 As DataContext, TArg1, TResult) ( _ 

 query As Expression(Of Func(Of TArg0, TArg1, TResult)) _ 

          ) As Func(Of TArg0, TArg1, TResult) 

 

 

Public Shared Function Compile(Of TArg0 As DataContext, TArg1, TArg2, TResult) ( _ 

 query As Expression(Of Func(Of TArg0, TArg1, TArg2, TResult)) _ 

          ) As Func(Of TArg0, TArg1, TArg2, TResult) 

 

 

Public Shared Function Compile(Of TArg0 As DataContext, TArg1, TArg2, TArg3, TResult) ( _ 

 query As Expression(Of Func(Of TArg0, TArg1, TArg2, TArg3, TResult)) _ 

          ) As Func(Of TArg0, TArg1, TArg2, TArg3, TResult) 

 

 

Public Shared Function Compile(Of TArg0 As DataContext, TResult) ( _ 

 query As Expression(Of Func(Of TArg0, TResult)) _ 

          ) As Func(Of TArg0, TResult) 

 

 

 و المحدد Compile عندما ٌتم تنفٌذ المفوض المعاد من الطرٌقة Delegate نوع المحدد الممرر للمفوض TArgحٌث تمثل المحددات 

TResult هو من النوع IEnumerable(T) المعاد عند تنفٌذ المفوض المعاد من الطرٌقة Compile 

فً العدٌد من الحالات ربما ترؼب فً إعادة استخدام الاستعلامات خارج مجال مسار التنفٌذ الحالً ففً هذه الحالات تكون عملٌة تخزٌن 

 مصممة من أجل Queries فكرة فعالة ففً الكود التالً توجد لدٌنا فئة Static Variablesالاستعلامات المترجمة فً متؽٌرات ساكنة 

  Linq to Sql بطرٌقة NorthWindتخزٌن الاستعلامات المترجمة تستخدم للاستعلام من قاعدة بٌانات 

Class Queries 

 

    Public Shared CustomersByCity As  _ 

        Func(Of NorthWindDataContext, String, IQueryable(Of Customer)) = _ 

            CompiledQuery.Compile(Function(db As NorthWindDataContext, _ 

            city As String) _ 

        From c In db.Customers Where c.City = city Select c) 

 

    Public Shared CustomersById As  _ 

        Func(Of NorthWindDataContext, String, IQueryable(Of Customer)) = _ 

            CompiledQuery.Compile(Function(db As NorthWindDataContext, _ 

            id As String) _ 

        db.Customers.Where(Function(c) c.CustomerID = id)) 

 

End Class 

 حٌث ٌمكننا تعرٌؾ وظٌفة تستخدم الاستعلام المترجم كما ٌلً

Public Function GetCustomersByCity(ByVal city As String) As  _ 

    IEnumerable(Of Customer) 

 

ms-help://MS.VSCC.v90/MS.msdnexpress.v90.en/fxref_system.data.linq/html/18dcdfc5-4021-bb96-6e55-86d4512d6552.htm
ms-help://MS.VSCC.v90/MS.msdnexpress.v90.en/fxref_system.data.linq/html/18dcdfc5-4021-bb96-6e55-86d4512d6552.htm
ms-help://MS.VSCC.v90/MS.msdnexpress.v90.en/fxref_system.data.linq/html/18dcdfc5-4021-bb96-6e55-86d4512d6552.htm
ms-help://MS.VSCC.v90/MS.msdnexpress.v90.en/fxref_system.data.linq/html/18dcdfc5-4021-bb96-6e55-86d4512d6552.htm

        Return Queries.CustomersByCity(db, city) 

End Function 

  كما ٌلDataGridViewًثم استخدام هذه الوظٌفة لعرض نتائج الاستعلام فً 

Me.DataGridView1.DataSource = GetCustomersByCity("London").ToList 

 DataGridViewأو حتى الاستخدام المباشر للاستعلام المترجم بحٌث نظهر نتٌجة الاستعلام مباشرة فً 

Me.DataGridView1.DataSource = Queries.CustomersByCity(db, "London").ToList 

